DOCUMENT RESUME

ED 052 610 EM 009 064

TITLE	BASIC Application Programs. Set 1 (15 Programs); Set 2 (19 Programs); Set 3 (12 Programs).
INSTITUTION	Digital Equipment Corp., Maynard, Mass.
PUB DATE	71
NOTE	Set 1, 33p.; Set 2, 42p.; Set 3, 26p.
AVAILABLE FROM	Digital Equipment Corporation, Educational Marketing (5-2), 146 Main St., Maynard, Mass. 10754 (set 1, $1.00; Set 2, $1.25; Set 3, $0.75)
EDRS PRICE	EDRS Price MF-$0.65 HC Not Available from EDRS.
DESCRIPTORS	Algebra, Arithmetic, Biology, *Business Subjects, Chemistry, *Computer Assisted Instruction, *Computer Programs, Concept Teaching, Ecology, Electronic Data Processing, Electronics, Games, Geometry, Mathematics, Physics, Program Descriptions, Risk, *Sciences, Social Studies, Statistics

ABSTRACT

The programs in these three booklets are designed to demonstrate how the computer can be applied to the problems of many disciplines. The problems and the corresponding computer programs are, for the most part, quite simple and are designed to be "jumping-off points" for students from the high school level up. Programs include one or more from each of the following areas: mathematics (calculating, algebra, geometry, plotting), statistics, physics, chemistry, biology, ecology, social studies, electricity and electronics, business, and games (slot machine or one-armed bandit). In addition to the computer program, the source, a description of the program, a discussion of the approach used in solving it, a sample printout, and the educational benefits are listed for each problem. The language BASIC is used for all programs. (Author/JK)

DIGITAL EQUIPMENT CORPORATION

education

BASIC APPLICATION PROGRAMS

INTRODUCTION

The programs contained in this book are designed to demonstrate how the computer can be applied in a meaningful way to problems of many disciplines. The problems and the corresponding programs are, for the most part, quite simple and are designed to be "jumping off points" for students from the high school level on up. Programs include one or more from each of the following areas:

- Mathematics
  - Calculating
  - Algebra
  - Geometry
  - Plotting
- Physics
- Chemistry
- Biology
- Ecology
- Social Studies
- Economics
- Business
- Electricity & Electronics
- Teacher Aids
- Games & Recreation

All of the programs, with a very few exceptions, may be run on EduSystems 10 through 90, PDP-8 Family and PDP-11 Family computers. Very large programs will not always run on EduSystems 10 (4K PDP-8); this is noted in the program description.

FOR MORE INFORMATION

For more advanced or comprehensive problems in BASIC, the following books are suggested:

Higgins, G. Albert, Jr.
The Elementary Functions: An Algorithmic Approach
Kiewit Computation Center
Dartmouth College
Hanover, New Hampshire 03755

Albrecht, R. & Finkel, L.
Math I, Math II, Business I and Social Studies I
Tecnica Education Corporation
655 Skyway Boulevard
San Carlos, California 97040

Sage, Edwin R.
Problem Solving With the Computer
Entelek, Incorporated
42 Pleasant Street
Newburyport, Massachusetts 01950

Johnson, D.C., et al
Computer Assisted Mathematics Program (CAMP) - 7 Volumes
Scott, Foresman & Company
Glenview, Illinois 60029

Coan, James S.
Basic BASIC
Hayden Book Company
New York, N.Y.

Albrecht, R., Lindberg, E., Mara, W.
Computer Methods in Mathematics
Addison-Wesley Publishing Company
Menlo Park, California 94025

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: DRINKR
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program illustrates how the computer, used as an extended calculator, can solve the following problem:

35 persons per 1000 have high blood pressure. 80% of those with high blood pressure drink, and 60% without high blood pressure drink. Estimate the fraction of drinkers with high blood pressure.

APPROACH: This problem requires the solution of several simple equations. If we let H = the number of people with high blood pressure, then:

$$H = 35$$

And if H1 = the number of people with high blood pressure who drink, then:

$$H1 = .80 * H$$

Letting L1 - the number of people with low blood pressure who drink yields:

$$L1 = .60 * (1000 - H)$$

Now, we can solve for the number of drinkers, D:

$$D = H1 + L1$$

Finally, the percentage of drinkers with high blood pressure is:

$$X = H1 * 100 / D$$

The program below solves the problem in a jiffy:

```
READY

 5 PRINT "PROGRAM DRINKR"
 6 PRINT
10 LET H=35
20 LET H1=.8*H
30 LET L1=.6*(1000-H)
40 LET D=H1+L1
50 LET X=H*100/D
60 PRINT X;"% DRINKERS HAVE HIGH BLOOD PRESSURE."
70 PRINT H*100/1000;"% OF ALL PEOPLE HAVE HIGH BLOOD PRESSURE."
99 END

RUN

PROGRAM DRINKR

 5.766063 % DRINKERS HAVE HIGH BLOOD PRESSURE.
 3.5 % OF ALL PEOPLE HAVE HIGH BLOOD PRESSURE.


READY
```

EDUCATIONAL BENEFITS: This type of problem can be written directly in BASIC without going through the steps above. Recognizing this type of problem readily will save lots of pencil-pushing time.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: Bicycl
SOURCE: Digital Equipment Corp LANGUAGE: BASIC
DESCRIPTION: The solution to this problem illustrates how the computer can be used as a powerful calculator. The problem is:

A boy on a bicycle and a man in a car start at the same time from town A for town B, 110 miles away. They travel over the same road at 6 and 40 miles per hour respectively. When the man in the car reaches B, he will stop for 15 minutes and then start back again. How many hours will the boy on the bicycle have traveled when he meets the car on its return trip?

USAGE: Let's set up our equations first. Remember distance = rate * time or T = D/R. Letting T1 be the time the car has traveled before it starts the return trip, then:

$$T1 = \frac{110 \text{ mi}}{40 \text{ mph}} + \frac{1}{4} \text{ hr.}$$

The bicycle has traveled T1 * 6 mph at the time the car starts back. Hence the time until they meet is

$$T2 = \frac{110 - T1*6}{40 \text{ mph} + 6 \text{ mph}}$$

The total time is T1 + T2, so now we can easily write a program to solve the problem:

```
 5 PRINT "PROGRAM BICYCL"
 6 PRINT
10 LET T1=110/40+1/4
20 LET T2=(110-6*T1)/(40+6)
30 PRINT "CYCLIST HAS TRAVELED";T1+T2;"HOURS WHEN BOTH MEET."
99 END

RUN

PROGRAM BICYCL

CYCLIST HAS TRAVELED 5 HOURS WHEN BOTH MEET.
```

Not very difficult and the numbers "come out even." But they don't have to. Let's try it over a distance of 112 miles with speeds of 7.5 mph and 46 mph respectively. The computer solves this one just as easily. Can you?

```
READY

10 LET T1=112/46+1/4
20 LET T2=(112-7.5*T1)/(46+7.5)

RUN

PROGRAM BICYCL

CYCLIST HAS TRAVELED 4.401869 HOURS WHEN BOTH MEET.


READY
```

EDUCATIONAL BENEFITS: Computers can be used as powerful calculators too.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: GROUP
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC
DESCRIPTION: This program illustrates a method of solving a mathematical word problem by trial and error using the computer. The problem is:

When 15 girls leave a group of boys and girls, there are 2 boys for each girl. After that, 45 boys leave. Then there are 5 girls for each boy. How many girls were in the group before anyone left?

USAGE: One must recognize that this problem requires the solution of two simultaneous equations. Letting G = the original number of girls and B = the original number of boys, the equations are:

$$(G - 15) \times 2 = B$$

$$(B - 45) \times 5 = (G - 15)$$

It is, of course, possible to substitute the value of B from Equation 1 into Equation 2 and solve for G. However, the computer permits us to get the solution by trial and error, obviously not a very sensible way to do the problem by hand.

The program below has been written to try all possible values of B and G between 1 and 100 until a solution is found or until the program runs out of combinations.

```
READY

 5 PRINT "PROGRAM GROUP"
 6 PRINT
10 FOR G=1 TO 100
20 FOR B=1 TO 100
25 LET I=I+1
30 IF 2*(G-15)<>B THEN 50
40 IF 5*(B-45)<>(G-15) THEN 50
45 PRINT G;"GIRLS",B;"BOYS"
48 GO TO 90
50 NEXT B
60 NEXT G
70 PRINT "NO SOLUTION"
90 PRINT I;"COMBINATIONS TRIED"
99 END

RUN

PROGRAM GROUP

 40 GIRLS       50 BOYS
 3950 COMBINATIONS TRIED


READY
```

EDUCATIONAL BENEFITS: The student can see that the computer can be employed to find a solution to a problem by trial and error, but that it must try 3950 possible solutions before the correct one is reached. Hence, if a problem can be easily solved by hand, the computer may not always be a significant advantage.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: SIMUL
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC
DESCRIPTION: The SIMUL program illustrates one way in which the computer can be used to solve relatively complex simultaneous equations.

USAGE: The computer can be used to solve simultaneous equations by trial and error far faster than we humans can. For example, let's solve for x and y in the following equations:

$$2^x = \frac{16y}{3} \quad \text{and} \quad 3^x = 27y$$

The program, which tries values of x and y between 0 and 100 until it reaches a solution or runs out of values is as follows:

```
  5 PRINT "PROGRAM SIMUL"
  6 PRINT "SOLUTION TO TWO SIMULTANEOUS EQUATIONS"
  7 PRINT
 10 FOR X=1 TO 100
 20 FOR Y=1 TO 100
 30 IF 2↑X<>16*Y/3 THEN 50
 40 IF 3↑X<>27*Y THEN 50
 45 PRINT "X =";X,"Y =";Y
 48 GO TO 99
 50 NEXT Y
 60 NEXT X
 70 PRINT "NO INTEGER SOLUTION"
 99 END

READY

RUN

PROGRAM SIMUL
SOLUTION TO TWO SIMULTANEOUS EQUATIONS

X = 4          Y = 3


READY
```

Let's try the same problem, but change Equation 2 slightly.

```
10 FOR X=1 TO 10
20 FOR Y=1 TO 10
40 IF 3*X<>28*Y THEN 50

RUN

PROGRAM SIMUL
SOLUTION TO TWO SIMULTANEOUS EQUATIONS

NO INTEGER SOLUTION


READY
```

Oh, Oh. What went wrong?

EDUCATIONAL BENEFITS: The student can see that trial and error, at least incrementing by whole numbers, does not always yield a solution, even trying 10,000 combinations. There must be another, better way.

Hint: Try combining the two equations and getting a solution for x using program ROOT. Then y can be easily solved for.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: BOOKS
SOURCE: Digital Equipment Corporation LANGUAGE: BASIC
DESCRIPTION: This program illustrates the solution to a fairly simple word problem involving two simultaneous equations. Then, by combining the equations and trying fewer combinations, the speed of obtaining the solution can be improved 100 fold. The problem is:

Brown has 48 books; some are worth $3 each, some $5 and others $8. The total value of the books is $175. How many of each kind are there if there are an even number of $5 books?

USAGE: The equations for solution are (letting T = number of three dollar books; F = five dollar books and E = eight dollar books):

$$T + F + E = 48$$

$$T * 3 + F * 5 + E * 8 = 175$$

If we then try all possible combinations of T, F, and E from 1 to 48 for each variable, our program would be as follows:

```
 5 PRINT "ALL POSSIBLE SOLUTIONS FOR BOOKS PROBLEM"
 6 PRINT
 7 PRINT "T","F","E"
 8 PRINT
10 FOR T=1 TO 48
20 FOR F=1 TO 48
30 FOR E=1 TO 48
40 IF (T+F+E)<>48 THEN 70
50 IF (T*3+F*5+E*8)<>175 THEN 70
60 PRINT T,F,E
70 NEXT E
80 NEXT F
90 NEXT T
99 END
```

```
RUN

BOOKS          21:03          22-JAN-71


ALL POSSIBLE SOLUTIONS FOR BOOKS PROBLEM

T              F              E

 34             13             1
 37             8              3
 40             3              5

TIME:  7.31 SECS.

READY
```

However, 7.31 seconds is a long time for the computer to work. Let's solve for T in both equations by hand and solve the resulting equation by brute force on the computer:

$$F * 2 + E * 5 = 31$$

In this equation, we can also reduce our limits since we recognize that F cannot be greater than 31/2 = 15.5 and E cannot be greater than 31/5 = 6.2. Making the appropriate program modifications leads to:

```
10
20 FOR F=1 TO 15
30 FOR E=1 TO 6
40 IF (F*2+E*5)<>31 THEN 70
50
55 LET T=48-F-E
90

RUN

BOOKS          21:06          22-JAN-71


ALL POSSIBLE SOLUTIONS FOR BOOKS PROBLEM

T              F              E

 40             3              5
 37             8              3
 34             13             1

TIME:  0.16 SECS.

READY
```

Our final refinement is to step F in increments of 2 since we know it must be even.

```
5 PRINT "SOLUTIONS FOR BOOKS PROBLEM WITH F = EVEN NUMBER"
20 FOR F=2 TO 14 STEP 2

RUN

BOOKS          21:10          22-JAN-71


SOLUTIONS FOR BOOKS PROBLEM WITH F = EVEN NUMBER

T              F              E

 37             8              3

TIME:  0.06 SECS.

READY
```

EDUCATIONAL BENEFIT: Notice the enormous improvement in computing time required for a solution 7.31 seconds to .06 seconds. Brute force certainly is inefficient! Perhaps it's worthwhile to think through most problems, particularly big ones. The computer may be fast, but we just improved its performance by 100 times using a little common sense!

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: SETS-1
SOURCE: CAMP First Course, Scott, Foresman and Company
LANGUAGE: BASIC
DESCRIPTION: Program to determine the intersection of two sets of numbers.

USAGE: Two sets of numbers can be combined to yield a third set by the operation of intersection. The intersection of two sets A and B is the set that contains all elements that belong to both A and B. It does not contain any other elements. The intersection is usually written A∩B.

For example, if M = { 0,2,4,6 } and K = { 1,2,3,4}, then M∩K = { 2,4 } .

The computer can be instructed to find the intersection of two sets. A program for finding the intersection of set x = { 1,3,5,---19 } and y = { 2,5,8,---29 } is written below

```
 5 PRINT "PROGRAM SETS-1"
 6 PRINT
 7 PRINT "THE INTERSECTION OF SETS X AND Y IS:"
10 FOR X=1 TO 19 STEP 2
20 FOR Y=2 TO 29 STEP 3
30 IF X=Y THEN 70
40 NEXT Y
50 NEXT X
60 GO TO 99
70 PRINT X
80 GO TO 50
99 END

RUN

PROGRAM SETS-1

THE INTERSECTION OF SETS X AND Y IS:
 5
 11
 17


READY
```

Using the computer, complicated sets which would be tedious to do by hand may be examined by defining one or both sets in a data statement. In the following example, set y is the same as above, but set x is given in the DATA statement.

```
  5 PRINT "PROGRAM SETS-1"
  6 PRINT
  7 PRINT "THE INTERSECTION OF SETS X AND Y IS:"
 10 READ X
 20 FOR Y=2 TO 29 STEP 3
 30 IF X=Y THEN 70
 40 NEXT Y
 50 GO TO 10
 70 PRINT X
 80 GO TO 10
 90 DATA 2,3,8,9,14,15,20,21,26,27
 99 END

READY


RUN

PROGRAM SETS-1

THE INTERSECTION OF SETS X AND Y IS:
 2
 8
 14
 20
 26

OUT OF DATA IN LINE 10


READY
```

EDUCATIONAL BENEFITS: The student is able to see that if there is a numerical pattern in the sets which intersect, there is also a pattern in the resultant intersecting set. In the first example above, for instance, set x steps by 2 and set y steps by 3, hence the intersecting set steps by 2 x 3 = 6. More complicated sets can be examined by computer than can easily be done by hand.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: TICKET
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC
DESCRIPTION: This program introduces the concept of logical branching through the vehicle of a simple word problem:

At a PTA fair, they wish to dispose of an article valued at $18.00 by selling tickets. These tickets are numbered consecutively beginning with one. Each person drawing a ticket pays in cents an amount equal to the number on the ticket, if that number is 50 or less. If the number drawn is greater than 50, he pays just 25¢. How many tickets must be sold so the PTA exactly breaks even?

USAGE: Letting T = the total money collected and N = the number of tickets sold, we can set up a flow chart to calculate the total money collected after each additional ticket is sold and compare that sum to $18.00.



Having the flow chart puts us just one step away from the program and a solution to the problem.

```
 5 PRINT "PROGRAM TICKET"
 6 PRINT
10 LET T=0
15 FOR N=1 TO 2000
20 IF N<=50 THEN 40
25 LET T=T+25
30 GO TO 50
40 LET T=T+N
50 IF T>=1800 THEN 70
60 NEXT N
70 PRINT N;"TICKETS SOLD TO COLLECT EXACTLY $";T/100
99 END

RUN

PROGRAM TICKET

 71 TICKETS SOLD TO COLLECT EXACTLY $ 18


READY
```

EDUCATIONAL BENEFITS: This problem can be done longhand, in which case it is extremely tedious. On the other hand, most students using a short cut method will get the wrong answer (72). So we see how the computer using logical branching can help us solve a deceptively simple problem quickly and correctly.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: BASKT
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC
DESCRIPTION: This program illustrates how the computer can be used to solve a problem using exponential convergence. The problem is:

Holding a ball six feet from the floor and letting it drop, a basketball player noticed that the ball bounced back up only two-thirds of that height (four feet). On the second bounce, the ball rose to only two thirds of the second height (two feet, eight inches). The third bounce brought it to two-thirds of the previous height, and so on. What was the total distance the ball traveled--both up and down--before it came to rest?

USAGE: A diagram can be used to show how far the ball bounces each time.



We notice that after the 1st bounce, the ball has traveled (at floor level):

$$D1 + 2 * D2 \quad \text{or} \quad D1 + 2* (2/3 * D1)$$

and that after 2 bounces, the ball has traveled:

$$D1 + 2*D2 + 2*D3 = D1 + 2*(2/3*D1) + 2*(2/3 * 2/3 * D1)$$

Hence, with each additional bounce, the ball travels an additional distance of:

$$2*D1*(2/3)^n \quad \text{or} \quad 12*(2/3)^n$$

Now it is a simple matter to write a program which increments the total distance traveled and print out the results after each bounce.

```
1 PRINT "PROGRAM BASKT"
2 PRINT
3 PRINT "BOUNCES","DISTANCE"
4 PRINT
5 LET D=6
10 FOR X=1 TO 50
20 LET D=D+12*(2/3)↑X
30 PRINT X,D
40 NEXT X
99 END

RUN

PROGRAM BASKT

BOUNCES         DISTANCE

 1               14
 2               19.33333
 3               22.88889
 4               25.25926
 5               26.83951
 6               27.893
 7               28.59534
 8               29.06356
 9               29.3757
 10              29.5838
 11              29.72253
 12              29.81502
 13              29.87668

 32              29.99994
 33              29.99996
 34              29.99997
 35              29.99998
 36              29.99998
 37              29.99999
 38              29.99999
 39              29.99999
 40              29.99999
 41              29.99999
 42              29.99999
 43              29.99999
 44              29.99999
 45              29.99999
 46              29.99999
 47              29.99999
 48              29.99999
 49              29.99999
 50              29.99999

READY
```

EDUCATIONAL BENEFIT: The student can see, by means of a contemporary example, how a progression can be used to solve a problem. He also learns to solve a problem which would be quite difficult to do by hand, especially after the first few terms.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: PROGRS
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: Sometimes the computer is used to do a problem in an entirely different way than we humans might approach the same problem. For example, given the following tables of values:

X	1	2	3	4	etc.
Y	10	17	26	37	

What is the value of X when Y is 2602?

APPROACH: We noticed that the difference between successive values of Y increases by 2 each step; i.e., 7, 9, 11, etc. Therefore, if we stepped from Y = 10 to 2602 in steps of Z, the step itself would have to be incremented by 2 for each new step. If we write a program to do this as well as count the number of steps, X, we should be able to solve the problem.

```
 5 PRINT "PROGRAM PROGRS"
 6 PRINT
10 LET X=1
20 LET Z=7
25 LET Y=10
30 LET Y=Y+Z
40 LET Z=Z+2
50 LET X=X+1
60 IF Y<2602 THEN 30
80 PRINT "X =";X,"Y =";Y
99 END

READY

RUN

PROGRAM PROGRS

X = 49          Y = 2602


READY
```

EDUCATIONAL BENEFIT: It's often possible to solve a problem using the computer in a manner quite differently than one would use by hand. How would you have solved this problem not using the computer?

DIGITAL EQUIPMENT CORPORATION

education

AREA: Geometry NUMBER: NAME: LADDER
SOURCE: Digital Equipment Corporation LANGUAGE: BASIC
DESCRIPTION: The ladder program shows how the computer can be used to calculate the solution to the following word problem:

A ladder 25 feet long is placed so that its foot is 7 feet from the base of a building. When the top of the ladder slipped 4 feet down the side of the building, how far did the foot of the ladder slip?

USAGE: The diagrams below show the two ladder positions.





By the Pythagorean theorem we know that $a^2 + b^2 = c^2$, hence, the equations needed to solve the above problem are:

$$x = \sqrt{25^2 - 7^2}$$

$$y = \sqrt{25^2 - (x-4)^2}$$

and the amount of slippage of the base is:

$$z = y - 7$$

The program to do the problem is as follows:

```
READY

 5 PRINT "PROGRAM LADDER"
 6 PRINT
10 LET X=SQR(25*25-7*7)
20 LET Y=SQR(25*25-(X-4)*(X-4))
30 LET Z=Y-7
40 PRINT "LADDER BASE SLIPPED";Z;"FEET."
99 END

RUN

PROGRAM LADDER

LADDER BASE SLIPPED 8 FEET.


READY
```

EDUCATIONAL BENEFITS: While the arithmetic in the problem above is not particularly "messy", it still is no great joy to solve by hand. However, the computer would be equally happy to do the problem if the ladder is 27.83 feet long and it was 7.62 feet from the wall originally. Would you?

```
10 LET X=SQR(27.83↑2-7.62↑2)
20 LET Y=SQR(27.83↑2-(X-4)↑2)
30 LET Z=Y-7.62

RUN

PROGRAM LADDER

LADDER BASE SLIPPED 8.386132 FEET.


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Geometry NUMBER: NAME: Area-1
SOURCE: Computer Methods in Mathematics LANGUAGE: BASIC

DESCRIPTION: The problem is to solve for the shaded area of the figure below for any value of the radius, R.



APPROACH: We start by remembering that the area of a circle and square are calculated as follows:

$$A_{circle} = \pi * R^2$$

$$A_{square} = SIDE^2 \text{ or } (2 * R)^2$$

The difference in area between a square and a circle inscribed within its borders is:

$$\Delta A = A_{square} - A_{circle}$$

and the area of one corner is:

$$A_{corner} = \Delta A/4$$

Now, let's write the program to perform this calculation for any input value of R.

EDUCATIONAL BENEFITS: By trying different values of R, the student can determine relationships in areas as a function of the components. Let's try several values of R and see what happens.

```
 5 PRINT "PROGRAM AREA-1"
 6 PRINT
 7 PRINT "WHAT IS THE RADIUS";
 8 INPUT R
 9 PRINT
10 LET P=3.14159
20 LET A1=P*R↑2
30 LET A2=(2*R)↑2
40 LET D=(A2-A1)/4
50 PRINT "AREA OF ENCLOSED CORNER IS";D
99 END

RUN

PROGRAM AREA-1

WHAT IS THE RADIUS? 1

AREA OF ENCLOSED CORNER IS .2146025


READY

RUN

PROGRAM AREA-1

WHAT IS THE RADIUS? 10

AREA OF ENCLOSED CORNER IS 21.46025


READY

RUN

PROGRAM AREA-1

WHAT IS THE RADIUS? 100

AREA OF ENCLOSED CORNER IS 2146.025


READY

RUN

PROGRAM AREA-1

WHAT IS THE RADIUS? 2.158652

AREA OF ENCLOSED CORNER IS 1


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Geometry NUMBER: NAME: CRSCNT
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: The computer is used to aid in the solution of the following problem. A crescent figure is bounded by a semicircle ABC whose radius is 5 inches and an arc ADC, the radius of whose circle is 10 inches. Find the area of the crescent.



APPROACH: In this problem, like any in which the computer is used, the student must recognize the fundamental formulae and algorithms to be used. If he cannot do that, he is stuck. To aid us in completing this problem, it will be helpful to complete the drawing to show more of both circles.



R1 = radius of small circle = 5
R2 = radius of big circle = 10

Some things should now be apparent. In particular, since ABC is a semicircle, distance AEC is equal to 2*R1 = 10. Hence, triangle ACF is equilateral. Therefore, the area of wedge ADCF equals 1/6

the area of the large circle. Hence, the area of the crescent is:

$$\text{Area}_{\text{crescent}} = \text{Area ABCE} - \text{Area ADCE}$$

$$= \frac{1}{2} * \text{Area}_{\text{small circle}} - \left[\text{Area}_{\text{wedge}} - \text{Area}_{\text{triangle}}\right]$$

$$= \frac{1}{2} * \pi * R1^2 - \left[\frac{1}{6} * \pi * R1^2 - \frac{1}{2} * R2 * \sqrt{R2^2 - \left(\frac{R2}{2}\right)^2}\right]$$

We can now substitute 5 and 10 for R1 and R2, reduce the equations and solve. The typical math text would ask that the answer be left in radical form and not figured for π. After a good bit of manipulation, the student would finally get down to an answer of:

$$\text{Area}_{\text{crescent}} = 25 * \left(\sqrt{3} - \frac{\pi}{6}\right)$$

However, using the computer, we can get a final solution without a lot of hand reduction.

```
10 LET P1=3.14159
20 PRINT "AREA OF CRESCENT IN SQUARE INCHES IS";
30 PRINT 1/2*P1*5↑2-1/6*P1*10↑2+1/2*10*SQR(10↑2-5↑2)
99 END

READY

RUN

AREA OF CRESCENT IN SQUARE INCHES IS 30.21131


READY
```

EDUCATIONAL BENEFITS: The student can see that the computer won't help him much unless he knows the problem fundamentals and can "teach" the computer how to help him. He also learns that the computer is happy to grind through calculations that aren't very pleasant to do by hand. Of course, if he wants to reduce the equation as far as possible and then use the computer, that approach works too.

```
10 LET P1=3.14159
20 PRINT "AREA OF CRESCENT IN SQUARE INCHES IS";
30 PRINT 25*(SQR(3)-P1/6)
99 END

READY

RUN

AREA OF CRESCENT IN SQUARE INCHES IS 30.21131


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Statistics NUMBER: NAME: FLIP-1
SOURCE: Basic BASIC, Hayden Publishing Co. LANGUAGE: BASIC

DESCRIPTION: This program makes use of the random number generator in BASIC to flip an imaginary coin and do a simple tabulation of the number of heads which come up. Ten sequences of 50 flips are tabulated.

APPROACH: Since the random number generator, RND, produces numbers between 0 and 1, we must have a way of splitting the numbers in half. One way, of course, is to ask whether the random number generated is greater or less than .5. Another way, which illustrates the use of the integer function, is as follows. Multiply the random number generated by 2 and then take the largest integer value contained in the result. If 0, let it be a tail; 1, a head. The statement to set the value of the flip is:

```
LET F = INT (2 * RND (1) )
```

Now, let's see what the whole program looks like.

```
  1 PRINT "PROGRAM FLIP-1"
  4 FOR Y=1 TO 10
  5 LET C=0
 10 FOR X=1 TO 50
 20 LET F=INT(2*RND(1))
 30 IF F=1 THEN 60
 40 PRINT "T";
 50 GO TO 100
 58 REM  C COUNTS THE NUMBER OF HEADS
 60 LET C=C+1
 70 PRINT "H";
100 NEXT X
110 PRINT
120 PRINT "HEADS";C;"OUT OF 50 FLIPS"
122 PRINT
125 NEXT Y
999 END
```

```
READY

RUN

PROGRAM FLIP-1
ITHTTTTTHTTTTHTHTTHHHTTTTHHHTTHHHHHTHHHHHHTTHHTTHT
HEADS 24 OUT OF 50 FLIPS

THTTHHHTHHTHHHHTTTHTHTHHTTHHHHHHTTTHHTHTTHTTTTTHTT
HEADS 25 OUT OF 50 FLIPS

THTTHTHHTTTHHHHHTHTHTTHHHTTHTHHTTHHTHHTTHTTHTHTHTT
HEADS 25 OUT OF 50 FLIPS

HTHTTTHTTTTHHHHHHHTHHTTHTTTTHHHHTHTTTTHTTHTHHTTTHT
HEADS 23 OUT OF 50 FLIPS

HHTHTHTTTHHTHTTTHTTTHTHTTTHHTTTTTTHTHTTHHHHTHTHHTT
HEADS 21 OUT OF 50 FLIPS

THHHTHHTTTHHTHTHTHHHTHTHTTTTTTTTHHTHHHHHTTHHTTTHTH
HEADS 25 OUT OF 50 FLIPS

THHTHTHHTHHTHHTHTHHHHHHTTHTTHHHHTHHTTTHTHHHHTTHHHT
HEADS 31 OUT OF 50 FLIPS

TTTHTTTTHTHHTTHTHTHTHTTHHTTHTTTHTHHHTHHHHTHHHHHHHH
HEADS 27 OUT OF 50 FLIPS

HHHHTHHHHHTHHHTHTHHTTHHHHTTHHTTHTHTHHTTTTTHTTTTHHH
HEADS 29 OUT OF 50 FLIPS

TTTHTHTTTTTTHTHTHTTTHTTHHTHTHHHTTHHHTHHHTHTHHHTHTT
HEADS 23 OUT OF 50 FLIPS
```

A simple addition lets us compute the percentage of all 500 flips which were heads.

```
2 LET C1=0
65 LET C1=C1+1
130 LET A=C1*100/500
133 PRINT
135 PRINT A;"% OF 500 FLIPS WERE HEADS"

RUN

PROGRAM FLIP-1
ITHTTTTTHTTTTHTHTTHHHTTTTHHHTTHHHHHTHHHHHHTTT
HEADS 24 OUT OF 50 FLIPS
                                        THHHHHHHH
THTTHHHTHHTHHHHTTTHTH
HEADS 25 OUT
                  THHHTTHHHHTTHHTTHTHTHHTTTTTHTTTTHHH
          OUT OF 50 FLIPS

ITTHTHHTTTTTHTHTHTTTHTTHHTHTHHHTTHHHTHHHTHTHHHTHTT
HEADS 23 OUT OF 50 FLIPS

 50.6 % OF 500 FLIPS WERE HEADS

READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business NUMBER: NAME: SORT-1
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: In business data processing, it is frequently necessary to sort a list of data into order. For example, some lists must be sorted by age, zip code, weight, inventory level, etc. This program is one way in which sorting can be done. The problem is to read in 10 numbers as DATA and then print them out sorted from smallest to largest.

APPROACH: In this type of problem, it is best to start with a flowchart.



```
READY

LIST
    2 DIM A(10),I(10)
    5 PRINT "RANDOMLY ORDERED DATA"
   10 FOR I=1 TO 10
   20 READ A(I)
   30 PRINT A(I);
   40 NEXT I
   50 LET C=0
   55 REM  CHECK IF ALL ELEMENTS IN LIST ARE ORDERED
   60 FOR I=1 TO 9
   70 IF A(I+1)<A(I) THEN 200
   80 NEXT I
   85 REM  C INDICATES IF AN ELEMENT WAS REORDERED
   90 IF C=0 THEN 300
  100 GO TO 50
  195 REM  REORDER ONE ELEMENT
  200 LET A=A(I)
  210 LET A(I)=A(I+1)
  220 LET A(I+1)=A
  230 LET C=1
  240 GO TO 80
  300 PRINT
  301 PRINT
  302 PRINT "ORDERED DATA"
  310 FOR I=1 TO 10
  320 PRINT A(I);
  330 NEXT I
  400 DATA 10,5,1,9,4,8,3,7,2,6
  999 END

READY

RUN

RANDOMLY ORDERED DATA
 10  5  1  9  4  8  3  7  2  6

ORDERED DATA
 1  2  3  4  5  6  7  8  9  10

READY



400 DATA 62,64,89,100,95,73,50,95,80,87

RUN

RANDOMLY ORDERED DATA
 62  64  89  100  95  73  50  95  80  87

ORDERED DATA
 50  62  64  73  80  87  89  95  95  100

READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Electricity NUMBER: NAME: Resist
SOURCE: Computer Methods in Mathematics LANGUAGE: BASIC

DESCRIPTION: The problem is to computer the equivalent resistance between points A and B for the circuits below:





APPROACH: We must remember that in series, resistance is additive, that is:

$$R_T = R_1 + R_2 \text{ ---}$$

Whereas in parallel, the resultant resistance is computed as follows:

$$\frac{1}{R_T} = \frac{1}{R_1} + \frac{1}{R_2} \text{ ---}$$

Given these relationships, it is a simple matter to write a program to compute the resultant resistance for the two circuits above.

```
 5 PRINT "PROGRAM RESIST"
 6 PRINT
10 PRINT "R1";
15 INPUT R1
20 PRINT "R2";
25 INPUT R2
30 PRINT "R3";
35 INPUT R3
40 LET R4=1/(1/R1+1/R2)+R3
50 PRINT
60 PRINT "R4 =";R4
99 END
```

```
RUN

PROGRAM RESIST

R1? 330
R2? 220
R3? 110

R4 = 242


READY

RUN

PROGRAM RESIST

R1? 10
R2? 10
R3? 10

R4 = 15


READY
```

For the second circuit, we need change only one statement in our program:

```
40 LET R4=1/(1/R1+1/R2+1/R3)

RUN

PROGRAM RESIST

R1? 330
R2? 220
R3? 110

R4 = 60


READY

RUN

PROGRAM RESIST

R1? 10
R2? 10
R3? 10

R4 = 3.333333


READY
```

EDUCATIONAL BENEFITS: The computer can be used as a tool to aid in solving all kinds of practical problems. It is not necessary for a student to get bogged down in mathematical calculations when he is eagerly learning about other things.

$1.25

DIGITAL EQUIPMENT CORPORATION

education

PROCESS WITH MICROFICHE AND PUBLISHER'S PRICES. MICRO-FICHE REPRODUCTION ONLY.

Basic Applications Programs (Set 2)

BASIC APPLICATION PROGRAMS

-Set 2-

The programs contained in this series are designed to demonstrate how the computer can be applied in a meaningful way to problems of many disciplines. The problems and the corresponding programs are, for the most part, quite simple and are designed to be "jumping off points" for students from the high school level on up.

All of the programs, with very few exceptions, may be run on EduSystems 10 through 90, PDP-8 family and PDP-11 family computers. Exceptions are noted in the program descriptions.

Additional copies of BASIC Applications Programs may be obtained from:

Direct Mail Department
Digital Equipment Corporation
Maynard, Massachusetts 01754

Price:	Set 1 (15 problems)	$1.00
	Set 2 (19 problems)	$1.25

BASIC APPLICATION PROGRAMS

Table of Contents (Sets 1 and 2)

Geometry	AREA-1	Solves for the area under a curve by equation.
	CRSCNT	Solves for the area of a crescent (not generalized).
	LADDER	Solves the slipping ladder problem by Pythagorean theorem.
Plotting	DIAMON	Plots an N by N diamond matrix.
	PLOTFN	Plots any function
	SINEX	Plots a sine wave
	3DPLOT	Plots any three dimension functions.
Statistics	FLIP-1	Flips a coin.
	GUESS	Binary search guessing game vs. the computer.
Business	DEPRECIATE	Demonstrates 3 methods of depreciation.
	ROUNDOFF	Rounds off numbers to any number of places.
	SORT-1	Sorts a list of numbers.
Biology	EVOLU	Demonstrates evolutionary mutations of pepper moths.
Physics	KINEMA	Tests knowledge of kinematics.
	LUNAR2	Simulates the landing of a lunar module on the moon.
	PHOTOE	Demonstrates photo electric effect.
	UELEC	Produces tables of electric potential.
Electronics	INDUC	Determines value of an unknown inductor.
	RESIST	Computes equivalent resistance for resistor networks.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: CONVRG
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC
DESCRIPTION: The CONVRG series of programs will calculate:

1. by an infinite series.
2. π by inscribed and circumscribe polygon.
3. π by infinite series (1 - 1/3 + 1/5 - 1/7 + 1/9...)

APPROACH: The above methods are commonly taught to show how and may be accurately calculated; however, it is difficult to carry out the calculations very far using pencil and paper alone.

The portion of the program to converge on e using the series

1 + 1/1 + 1/2 + 1/6 + 1/24 + 1/120 +...)

is as follows:

```
100 LET E=1
110 LET I=I+1
120 LET D=1
130 FOR J=1 TO I
140 LET D=D*J
150 NEXT J
160 LET E=E+1/D
170 PRINT E
180 GO TO 110
999 END

RUN

 2
 2.5
 2.666667
 2.708333
 2.716667
 2.718056
 2.718254
 2.718279
 2.718281
 2.718282
 2.718282
 2.718282
```

The portion of the program to converge on Pi by means of inscribed and circumscribed polygons is as follows:

```
 10 LET N=6
 20 LET N=2*N
 30 LET X=360/(N*57.29578)
 40 PRINT N*SIN(X)*COS(X)/2, N*TAN(X)/2
 50 GOTO 20
 99 END

READY

RUN

 2.598076        3.464102
 3               3.21539
 3.105829        3.15966
 3.132629        3.146086
 3.13935         3.142715
 3.141032        3.141873
 3.141453        3.141663
 3.141558        3.14161
 3.141584        3.141597
 3.141591        3.141594
 3.141592        3.141593
 3.141593        3.141593
 3.141593        3.141593

↑C
```

Convert degrees to radians.

Start with a 12-sided polygon

49,152-sided polygon

The third option, converging on Pi using an infinite series converges very slowly. Therefore, only every 500th value is typed out.

```
300 LET S=1
310 LET I=1
320 LET Q=Q+1
330 LET P=P+S/I
340 LET I=I+2
350 LET S=-S
360 IF Q<499 THEN 320
370 LET Q=0
380 PRINT P*4
390 GO TO 320
999 END

RUN

 3.143597
 3.14059
 3.142261
 3.141091
 3.141994
 3.141259
 3.141879
 3.141342
 3.141816
 3.141392
 3.141775
 3.141426
 3.141748

↑C
```

7 seconds for each new value.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: GOLDBACH
SOURCE: CAMP First Course, Scott, Foresman and Co. LANGUAGE: BASIC

DESCRIPTION: This is the description of the program exercise as it appears in the CAMP text (page 200).

"Goldbach's Conjecture is one of the famous unproved theories in mathematics. Many people have searched for a counter-example to prove that the following claim is false: Every even number E greater than four is the sum of two odd prime numbers. Study the program "Goldbach" which will print out each even number E, where $E<6<50$, and two odd prime numbers whose sum is E. Examine the output from the computer. Does this program "prove" the conjecture.

The teacher's guide is quick to point out that it proves the conjecture for $E \leq 50$, but that's all.

USAGE: Load Goldbach and type RUN. No input is required. This program has nested loops which take time. Running time is in minutes.

EDUCATIONAL BENEFITS: The student has whole new ways to approach theorem proving when the computer is available to try lots of cases.

```
LIST

   5 PRINT "GOLDBACH"
  10 FOR E=6 TO 50 STEP 2
  15 FOR P=3 TO E-3
  20 FOR F=2 TO P-1
  25 IF P/F=INT(P/F) THEN 65
  30 NEXT F
  35 FOR N=3 TO E-3
  40 FOR D=2 TO N-1
  45 IF N/D=INT(N/D) THEN 60
  50 NEXT D
  55 IF P+N=E THEN 80
  60 NEXT N
  65 NEXT P
  70 PRINT "CONJECTURE FALSE FOR"E
  75 GO TO 99
  80 PRINT E"="P"+"N
  85 NEXT E
  99 END

READY

RUN

GOLDBACH
 6 = 3 + 3
 8 = 3 + 5
 10 = 3 + 7
 12 = 5 + 7
 14 = 3 + 11
 16 = 3 + 13
 18 = 5 + 13
 20 = 3 + 17
 22 = 3 + 19
 24 = 5 + 19
 26 = 3 + 23
 28 = 5 + 23
 30 = 7 + 23
 32 = 3 + 29
 34 = 3 + 31
 36 = 5 + 31
 38 = 7 + 31
 40 = 3 + 37
 42 = 5 + 37
 44 = 3 + 41
 46 = 3 + 43
 48 = 5 + 43
 50 = 3 + 47


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: PASCAL
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program simulates the dropping of N balls through the following array of K levels. It is assumed that each ball is equally likely to fall to the left or right at each point in the array. After each group of balls has been dropped, the number of balls in each basket (from left to right) is typed out.

The number of balls landing in the various "baskets" at each level, when reduced to the lowest common denominator, should approximate the numbers in Pascal's Triangle. Do they? Write a program to determine how close you come. You should be able to observe many interesting things about Pascal's Triangle. What are the sums of the numbers in each row? What is the progression of numbers in each diagonal?

```
  5 PRINT "PROGRAM PASCAL"
  6 PRINT
 10 DIM B(10)
 15 PRINT "HOW MANY BALLS";
 16 INPUT M
 17 PRINT "HOW MANY LEVELS";
 18 INPUT K
 20 PRINT
 30 FOR N=1 TO M
 35 LET T=0
 40 FOR L=1 TO K
 50 IF RND(1)>.5 THEN 100
 90 LET T=T+1
100 NEXT L
110 LET B(T+1)=B(T+1)+1
120 NEXT N
130 FOR L=1 TO (K+1)
140 PRINT L,B(L)
145 LET B(L)=0
150 NEXT L
160 GOTO 6
999 END

READY
```





Pascal's Triangle

```
RUN

PROGRAM PASCAL

HOW MANY BALLS? 1000
HOW MANY LEVELS? 1

 1                506
 2                494

HOW MANY BALLS? 1000
HOW MANY LEVELS? 2

 1                252
 2                501
 3                247

HOW MANY BALLS? 1000
HOW MANY LEVELS? 3

 1                117
 2                393
 3                376
 4                114

HOW MANY BALLS? 1000
HOW MANY LEVELS? 4

 1                56
 2                250
 3                366
 4                254
 5                74

HOW MANY BALLS? 1000
HOW MANY LEVELS? 5

 1                30
 2                135
 3                321
 4                317
 5                160
 6                37

HOW MANY BALLS? 1000
HOW MANY LEVELS? 6

 1                16
 2                102
 3                225
 4                323
 5                226
 6                93
 7                15

HOW MANY BALLS?
↑C

READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: QUADRT
SOURCE: Digital Equipment Corporation LANGUAGE: BASIC
DESCRIPTION: For any input values a, b, and c of a first degree quadratic equation ($ax^2 + bx + c = 0$), this program computes the roots of the equation.

APPROACH: Based on the quadratic theorem, given $ax^2 + bx + c = 0$, then

$$x = \frac{-b \pm \sqrt{b^2 - 42c}}{2a}$$

Assuming a, b, and c are real numbers, the following principles are applied:

1. If $b^2 - 4ac$ is positive, then the roots are real and unequal.
2. If $b^2 - 4ac$ equals 0, then the roots are real and equal.
3. If $b^2 - 4ac$ is negative, then the roots are imaginary and unequal.

The program and several sample runs are on the reverse.

```
RUN

THIS PROGRAM SOLVES FOR THE ROOTS OF A
QUADRATIC EQUATION.  PLEASE INPUT THE FOLLOWING:
VALUE OF A? 1
VALUE OF B? 2
VALUE OF C? 1

THE ROOTS ARE
-1


THIS PROGRAM SOLVES FOR THE ROOTS OF A
QUADRATIC EQUATION.  PLEASE INPUT THE FOLLOWING:
VALUE OF A? 0
VALUE OF B? 2
VALUE OF C? 4

THIS IS A FIRST DEGREE EQUATION.
```

```
READY

10 PRINT "THIS PROGRAM SOLVES FOR THE ROOTS OF A "
20 PRINT "QUADRATIC EQUATION.  PLEASE INPUT THE FOLLOWING:"
22 PRINT "VALUE OF A";
24 INPUT A
26 PRINT "VALUE OF B";
28 INPUT B
30 PRINT "VALUE OF C";
32 INPUT C
34 PRINT
35 LET R=B↑2-4*A*C
40 IF A<>0 THEN 50
45 PRINT "THIS IS A FIRST DEGREE EQUATION."
47 GOTO 90
50 PRINT "THE ROOTS ARE ";
55 IF R<0 THEN 80
56 PRINT
58 IF R=0 THEN 70
60 PRINT (-B+SQR(R))/2*A
62 PRINT (-B-SQR(R))/2*A
65 GOTO 90
70 PRINT -B/2*A
75 GOTO 90
80 PRINT "IMAGINARY"
82 PRINT (-B/2*A);" + ";SQR(-R)/2*A;" *I"
84 PRINT (-B/2*A);" - ";SQR(-R)/2*A;" *I"
90 END

RUN

THIS PROGRAM SOLVES FOR THE ROOTS OF A
QUADRATIC EQUATION.  PLEASE INPUT THE FOLLOWING:
VALUE OF A? 2
VALUE OF B? 5
VALUE OF C? 2

THE ROOTS ARE
-2
-.5


READY

RUN

THIS PROGRAM SOLVES FOR THE ROOTS OF A
QUADRATIC EQUATION.  PLEASE INPUT THE FOLLOWING:
VALUE OF A? 4
VALUE OF B? 2
VALUE OF C? 4

THE ROOTS ARE IMAGINARY
-.4  +  1.49193  *I
-.4  -  1.49193  *I


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: ROOTS
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: ROOT finds all the roots of a function (ANY FUNCTION!) between -20 and 20. The function may be quadratic, cubic, trigonometric or any combination.

APPROACH: The method used here involves evaluating the function at incremental values, finding places where the value of the function changes sign and then, by successive approximations, finding the zero point. This method is similar to the commonly taught "Newton's Method."

USAGE: Before typing RUN, enter the function to be evaluated as a DEF FNA(X) statement as line 100. Then type RUN. For example:

```
100 DEF FNA(X) = 2*X↑3+11*X↑2-31*X-18Ø

100 DEF FNA(X) = SIN(X)-.5

100 DEF FNA(X) = X-4
```

EDUCATIONAL BENEFITS: The average student (especially in high school) does not have a thorough understanding of the very important concept of solving a function, much less an ability to do so for functions of any complexity. The computer lets them become very knowledgeable and confident in this area.

```
LIST

 100 DEF FNA(X)=2*X↑3 + 11*X↑2 - 31*X - 180
 110 LET Z1=-200
 120 FOR I=-19.9 TO 20
 130 IF SGN(FNA(I))=SGN(FNA(I+1)) THEN 300
 140 LET K=I
 150 LET J=I+1
 160 IF FNA(K)<FNA(J) THEN 200
 170 LET Z=K
 180 LET K=J
 190 LET J=Z
 200 LET Z=(K+J)/2
 210 IF FNA(Z)<0 THEN 240
 220 LET J=Z
 230 GO TO 250
 240 LET K=Z
 250 IF ABS(FNA(Z))>5.000000E-05 THEN 200
 260 LET Z=SGN(Z)*INT(ABS(Z)*10000+.5)/10000
 270 IF Z=Z1 THEN 300
 280 PRINT "F("Z")=0"
 290 LET Z1=Z
 300 NEXT I
 310 END

READY

100 DEF FNA(X)=X-4
RUN

F( 4 )=0


READY

100 DEF FNA(X)=SIN(X)-.5
RUN

F(-18.3259 )=0
F(-16.2315 )=0
F(-12.0428 )=0
F(-9.9483 )=0
F(-5.7596 )=0
F(-3.6651 )=0
F( .5236 )=0
F( 2.6179 )=0
F( 6.8068 )=0
F( 8.9011 )=0
F( 13.09 )=0
F( 15.1844 )=0
F( 19.3732 )=0


READY

100 DEF FNA(X)=2*X↑3 + 11*X↑2 - 31*X - 180
RUN

F(-5 )=0
F(-4.5 )=0
F( 4 )=0
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: TUTOR-1
SOURCE: Walter Koetke, Lexington High School LANGUAGE: BASIC

DESCRIPTION: An effective technique for stimulating interest as well as for teaching methods of solving word problems is to have students write a program which will create and present a problem, ask the user to solve the problem, and then verify the user's response. For example, consider the problem:

A car traveling C (have computer generate an integer 40 through 65) miles per hour can make a certain trip in D (have computer generate an integer 5 through 20) hours less than a train traveling at T (have computer generate an integer 20 through 39) miles per hour. How long does the trip take by car?

The student's program should type out this problem with numbers generated by the random function appropriately substituted, then ask for the user's solution, the student's program must also correctly solve the problem. A program which does all of this is:

```
 5 RANDOMIZE
10 LET C=INT(RND(0)*25)+40
15 LET D=INT(RND(0)*15)+5
20 LET T=INT(RND(0)*19)+20
25 PRINT "A CAR TRAVELING"C"MPH CAN MAKE A CERTAIN TRIP IN"
 30 PRINT D"HOURS LESS THAN A TRAIN TRAVELING AT"T"MPH."
 35 PRINT "HOW LONG DOES THE TRIP TAKE BY CAR";
40 INPUT A
45 LET V=D*T/(C-T)
50 LET E=INT(ABS((V-A)*100/A)+.5)
55 IF E>5 THEN 70
60 PRINT "GOOD!  ANSWER WITHIN"E"PERCENT."
65 GO TO 80
70 PRINT "SORRY.  YOU WERE OFF BY"E"PERCENT."
80 PRINT "CORRECT ANSWER IS"V"HOURS."
99 END

READY
```

```
RUN

A CAR TRAVELING 64 MPH CAN MAKE A CERTAIN TRIP IN
 12 HOURS LESS THAN A TRAIN TRAVELING AT 21 MPH.
HOW LONG DOES THE TRIP TAKE BY CAR? 6
GOOD!  ANSWER WITHIN 2 PERCENT.
CORRECT ANSWER IS 5.860465 HOURS.


READY

RUN

A CAR TRAVELING 53 MPH CAN MAKE A CERTAIN TRIP IN
 7 HOURS LESS THAN A TRAIN TRAVELING AT 23 MPH.
HOW LONG DOES THE TRIP TAKE BY CAR? 5
SORRY.  YOU WERE OFF BY 7 PERCENT.
CORRECT ANSWER IS 5.366667 HOURS.


READY

RUN

A CAR TRAVELING 43 MPH CAN MAKE A CERTAIN TRIP IN
 19 HOURS LESS THAN A TRAIN TRAVELING AT 27 MPH.
HOW LONG DOES THE TRIP TAKE BY CAR? 32
GOOD!  ANSWER WITHIN 0 PERCENT.
CORRECT ANSWER IS 32.0625 HOURS.


READY

RUN

A CAR TRAVELING 45 MPH CAN MAKE A CERTAIN TRIP IN
 7 HOURS LESS THAN A TRAIN TRAVELING AT 22 MPH.
HOW LONG DOES THE TRIP TAKE BY CAR? 7
GOOD!  ANSWER WITHIN 4 PERCENT.
CORRECT ANSWER IS 6.695652 HOURS.


READY
```

EDUCATIONAL BENEFIT: An effective use of this type of problem is to assign one problem from a list of several different problems to each student or to small groups of students. After the programs are written, students are directed to check programs written by others. In this way, each student not only fully understands the type of problem assigned to him, but he also gets practice in solving all of the other types as well. Note that the student must solve the problem in general in ord to write his program. This should give him more valuable experience th olving two or three similar problems with specific numeric values.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Plotting NUMBER: NAME: DIAMON
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program prints an N by N matrix of diamond shapes, each one being R (odd number input) characters wide.

```
 3 PRINT "FOR A PRETTY DIAMOND PATTERN, "
 4 PRINT "TYPE IN AN ODD NUMBER BETWEEN 4 AND 32."
 5 INPUT R
 6 LET Q=INT(60/R)
 8 FOR L=1 TO Q
11 LET X=1
12 LET Y=R
13 LET Z=2
20 FOR N=X TO Y STEP Z
25 PRINT TAB((R-N)/2);
28 FOR M=1 TO Q
29 LET C=1
30 FOR A= 1 TO N
32 IF C=1 THEN 40
33 IF C=2 THEN 42
34 IF C=3 THEN 44
35 PRINT "!";
37 GOTO 50
40 PRINT "D";
41 GOTO 47
42 PRINT "E";
43 GO TO 47
44 PRINT "C";
47 LET C=C+1
50 NEXT A
53 IF M=Q THEN 60
55 PRINT TAB(R*M+(R-N)/2);
56 NEXT M
60 PRINT
70 NEXT N
83 IF X<>1 THEN 95
85 LET X=R-2
86 LET Y=1
87 LET Z=-2
90 GOTO 20
95 NEXT L
99 END
```

```
FOR A PRETTY DIAMOND PATTERN,
TYPE IN AN ODD NUMBER BETWEEN 4 AND 32.
? 15
```



DIGITAL EQUIPMENT CORPORATION

education

AREA: Plotting NUMBER: NAME: PLOTFN
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program accepts any input function, f(x), at the time of program execution and plots that function from X_{min} to X_{max} in steps of S. The program also inserts the y axis, if it is crossed. Functions which can be plotted range from very simple to rather complex. All of the following are plottable functions:

Function	X Limits		Increment
f(x) = 2 * x	-15	+15	1
f(x) = 30*SIN(x)	-5	+5	.25
f(x) = x-x↑2	-5	+6	.5
f(x) = 30*SIN(.5*x)*EXP(-.2*ABS(x))	0	+15	.25

The program is on the reverse. A sample plot is shown below:

```
WHAT ARE THE INITIAL AND FINAL VALUES OF X, AND THE
INCREMENT SIZE? -5,6,1
```



```
110 PRINT "TO PLOT Y=F(X), THE USER MUST TYPE:"
120 PRINT
130 PRINT "100 GOTO 200"
140 PRINT "200 DEF FNA(X)=(THE EXPRESSION FOR F(X))"
150 PRINT "RUN"
160 PRINT
190 STOP
200 DEF FNA(X)=2*X
210 PRINT
220 PRINT "WHAT ARE THE INITIAL AND FINAL VALUES OF X, AND THE"
 230 PRINT "INCREMENT SIZE";
240 INPUT X1, X2, S
250 PRINT
260 PRINT
270 PRINT TAB(33); "X ="; X1
280 FOR X=X1 TO X2 STEP S
290 IF ABS(X)<.000001 THEN 362
340 LET Y = FNA(X) + 35
350 IF Y>35 THEN 380
360 PRINT TAB(Y);"*";TAB(35);"."
361 GO TO 390
362 PRINT "-Y . . . . . .";
363 PRINT " . . . . . +Y";TAB(FNA(X)+35);"*"
364 PPINT TAB(4);"-30";TAB(14);"-20";TAB(24);"-10";TAB(35);"0";
 365 PRINT TAB(44);"+10";TAB(54);"+20";TAB(64);"+30";
370 GOTO 390
380 PRINT TAB(35);".";TAB(Y);"*"
390 NEXT X
400 PRINT TAB(33);"X =";X2
410 PRINT
440 END

READY

RUN

TO PLOT Y=F(X), THE USER MUST TYPE:

100 GOTO 200
200 DEF FNA(X)=(THE EXPRESSION FOR F(X))
RUN


READY

100 GOTO 200
200 DEF FNA(X)=X↑2-X
RUN


WHAT ARE THE INITIAL AND FINAL VALUES OF X, AND THE
INCREMENT SIZE? -5,6,1
```

(See Plot on Front of Page)

DIGITAL EQUIPMENT CORPORATION

education

AREA: Plotting NUMBER: NAME: SINEX
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: The computer can be used to plot virtually any function. This particular method takes advantage of the TAB command, although it need not. For example, a program to plot a sine wave without using TAB, would be as follows:

```
 1 FOR X=0 TO 19 STEP .25
 2 LET Q=30+30*SIN(X)
 3 FOR B=1 TO Q
 4 PRINT " ";
 5 NEXT B
 6 PRINT "X"
 7 NEXT X
 9 END
```

In the next program, we have filled in the area between the y axis and the curve and we have used the TAB command to replace steps 3 to 5 above.

```
31 FOR X=0 TO 19 STEP .25
32 LET Q=30*SIN(X)*EXP(-.1*X)
33 IF Q>=0 THEN 35
34 GO TO 42
35 PRINT TAB(30);
36 FOR Y=0 TO Q
37 PRINT "X";
38 NEXT Y
40 PRINT
41 GO TO 50
42 PRINT TAB(30+Q);
43 FOR Y=0 TO Q STEP -1
44 PRINT "X";
45 NEXT Y
46 PRINT
50 NEXT X
99 END
```

RUN



DIGITAL EQUIPMENT CORPORATION

education

AREA: Plotting NUMBER: NAME: 3DPLOT
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: 3DPLOT will plot a three dimensional, perspective graph of a function of three variables. It plots Z (the vertical axis) as a function of X and Y, where X and Y are inside a circle of radius 30.

USAGE: Enter the function to be plotted as a DEF FNA(Z) statement in line 5. Then type RUN. The resulting plot will go sideways on the page.

```
  5 DEF FNA(Z)=SQR(900.01-Z*Z)*.9-2
 40 PRINT
 45 LET I1=1
 50 FOR X=-30 TO 30 STEP 1.5
 60 LET L1=-1
 70 LET Y1=5*INT(SQR(900.01-X*X)/5)
 80 FOR Y=Y1 TO -Y1 STEP -5
 90 LET Z=INT(25+FNA(SQR(X*X+Y*Y))-.7*Y)
100 IF L1=-1 THEN 140
110 IF Z>L1 THEN 180
120 IF Z<L2 THEN 150
130 GO TO 200
140 LET L1=Z
150 LET L2=Z
160 IF Z=I1-1 THEN 166
161 FOR I=I1 TO Z
162 PRINT " ";
164 NEXT I
166 PRINT "*";
167 LET I1=Z+2
170 GO TO 200
180 LET L1=Z
190 GO TO 160
200 NEXT Y
210 PRINT
215 LET I1=1
220 NEXT X
230 PRINT
240 END

READY
```

Like all plotting programs, some functions come out "prettier" than others. Two that are known to work are:

```
5 DEF FNA(Z) = 30*EXP(-Z*Z/100)

5 DEF FNA(Z) = SQR (9ØØ.Ø1-Z*Z)*.9 -2
```

```
5 DEF FNA(Z)=30*EXP(-Z*Z/100)

RUN
```



```
READY
```

EDUCATIONAL BENEFITS: Using the computer to plot is an excellent way to teach coordinate geometry. It really shows the interrelation of geometry and algebra as described by Descartes.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Statistics/Games NUMBER: NAME: GUESS
SOURCE: Walter Koetke, Lexington High School LANGUAGE: BASIC

DESCRIPTION: The technique of binary search has a number of mathematical and programming applications. In program GUESS, the computer chooses a random integer between 0 and 100 and the student must try to guess what it is in as few tries as possible.

Note that the program does not reveal the technique of binary search, but merely alludes to the fact that no more than 7 guesses are necessary. Most students "discover" the idean of binary search after only a few runs of the program.

```
10 PRINT "I HAVE CHOSEN AN INTEGER 0 THROUGH 100.  TRY TO"
15 PRINT "GUESS MY NUMBER IN AS FEW TRIES AS POSSIBLE."
16 PRINT
20 LET C=0
22 RANDOMIZE
25 LET N=INT(RND(0)*100)
30 LET C=C+1
35 PRINT "YOUR GUESS IS";
40 INPUT G
45 IF N=G THEN 70
50 IF G>N THEN 60
55 PRINT "TOO LOW"
56 GO TO 30
60 PRINT "TOO HIGH"
65 GO TO 30
70 PRINT "CORRECT IN"C"GUESSES."
75 IF C>3 THEN 90
80 PRINT "YOU WERE LUCKY!"
85 STOP
90 IF C>7 THEN 100
95 PRINT "GOOD JOB."
98 STOP
100 PRINT "BUT YOU SHOULDN'T NEED MORE THAN 7 GUESSES."
105 END
```

```
READY

RUN

I HAVE CHOSEN AN INTEGER Ø THROUGH 1ØØ.  TRY TO
GUESS MY NUMBER IN AS FEW TRIES AS POSSIBLE.

YOUR GUESS IS? 5Ø
TOO HIGH
YOUR GUESS IS? 25
TOO LOW
YOUR GUESS IS? 37
CORRECT IN 3 GUESSES.
YOU WERE LUCKY!


READY

RUN

I HAVE CHOSEN AN INTEGER Ø THROUGH 1ØØ.  TRY TO
GUESS MY NUMBER IN AS FEW TRIES AS POSSIBLE.

YOUR GUESS IS? 4Ø
TOO LOW
YOUR GUESS IS? 8Ø
TOO HIGH
YOUR GUESS IS? 6Ø
TOO HIGH
YOUR GUESS IS? 66
TOO HIGH
YOUR GUESS IS? 45
TOO HIGH
YOUR GUESS IS? 35
TOO LOW
YOUR GUESS IS? 41
TOO LOW
YOUR GUESS IS? 43
TOO LOW
YOUR GUESS IS? 44
CORRECT IN 9 GUESSES.
BUT YOU SHOULDN'T NEED MORE THAN 7 GUESSES.


READY
```

EXTENSIONS: An excellent student assignment is writing a program which reverses the role of user and computer in Program GUESS. The computer should be programmed to guess a number the user has determined. After each guess, the user inputs 1, 2 or 3 for LOW, HIGH or CORRECT and the program continues until the correct number is found.

The program should guess any number between 0 and 100 in 7 guesses provided all the user clues are consistent. Writing the steps necessary to check the consistency of clues can be part of an assignment for better students.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business NUMBER: NAME: DEPRECIATE
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program shows how a piece of capital equipment depreciates according to three commonly used depreciation methods: straight line, sum of the digits, and double declining.

```
120 PRINT "ORIGINAL COST";
130 INPUT C
140 PRINT " LIFE OF ITEM";
150 INPUT L
160 PRINT "  SCRAP VALUE";
170 INPUT S
180 PRINT
190 PRINT "YEAR","STRAIGHT","SUM OF"," DOUBLE"
200 PRINT " "," LINE","DIGITS","DECLINING"
210 PRINT
220 LET V=C-S
230 LET D1=V/L
240 LET Y=((L+1)/2)*L
250 LET Z=L
260 FOR X=1 TO L
270 LET D2=V*(Z/Y)
280 LET Z=Z-1
290 LET D3=2*C/L
300 LET C=C-D3
310 PRINT X,
312 LET Q=D1
314 GOSUB 400
316 LET Q=D2
318 GOSUB 400
320 LET Q=D3
322 GOSUB 400
325 PRINT
330 NEXT X
350 STOP
400 LET Q=INT(Q*100)/100
420 IF Q>100 THEN 440
430 PRINT " ";
440 IF Q>10 THEN 460
450 PRINT " ";
460 PRINT "$"Q,
490 RETURN
999 END
```

USAGE: Load the DEPRECIATE program and type RUN. Enter the purchase price of the item, its life in years, and its scrap value at the end of that time.

```
RUN

ORIGINAL COST? 1000
 LIFE OF ITEM? 5
  SCRAP VALUE? 100

YEAR          STRAIGHT LINE     SUM OF DIGITS     DOUBLE DECLINING

 1            $ 180             $ 300             $ 400
 2            $ 180             $ 240             $ 240
 3            $ 180             $ 180             $ 144
 4            $ 180             $ 120             $ 86.4
 5            $ 180             $ 60              $ 51.84

READY

RUN

ORIGINAL COST? 2000
 LIFE OF ITEM? 4
  SCRAP VALUE? 300

YEAR          STRAIGHT LINE     SUM OF DIGITS     DOUBLE DECLINING

 1            $ 425             $ 680             $ 1000
 2            $ 425             $ 510             $ 500
 3            $ 425             $ 340             $ 250
 4            $ 425             $ 170             $ 125

READY
```

EDUCATIONAL BENEFITS: The study of "general business" is often just the study of a lot of tables in a textbook. It's hard for a student to get involved in this work, since he often doesn't understand what is behind these tables. The computer can bring many of these topics to life.

Also, by having some exposure to a computer, the student learns to be comfortable with it. He will be much better prepared for work in a modern business.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business/Math NUMBER: NAME: ROUNDOFF
SOURCE: F. McPhetres, Hartford High School LANGUAGE: BASIC

DESCRIPTION: Frequently, it is desirable to round off numbers to 2 decimal places (cents), 0 decimal places (dollars), -2 decimal places (hundreds of dollars), etc. If you wish to round off a decimal number in a program of your own, simply insert the following line:

```
LET R = INT(W*10↑Y+0.5)/(10↑Y)
```

W is the number you wish to round off (or a variable which represents the number) while Y is the number of decimal places desired. That is, Y is 3 if you wish to round to the nearest thousandth, 2 to round to hundredths, 0 to round to the nearest whole number, -1 to round to the nearest multiple of 10, etc.

Below is a program whose sole function is to round off decimal numbers.

```
100 PRINT "WHAT NUMBER DO YOU WISH TO ROUND OFF";
110 INPUT N
120 PRINT "TO HOW MANY PLACES";
130 INPUT Y
140 PRINT
150 LET A=INT(N*10↑Y+0.5)/(10↑Y)
160 PRINT N "=" A "TO" Y "DECIMAL PLACES."
170 PRINT
180 GOTO 100
190 END
```

```
RUN

WHAT NUMBER DO YOU WISH TO ROUND OFF? 56.0237
TO HOW MANY PLACES? 2

 56.0237 = 56.02 TO 2 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 8.449
TO HOW MANY PLACES? 1

 8.449 = 8.4 TO 1 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 3.685
TO HOW MANY PLACES? 2

 3.685 = 3.68 TO 2 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 3.67499
TO HOW MANY PLACES? 2

 3.67499 = 3.67 TO 2 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 3.67501
TO HOW MANY PLACES? 2

 3.67501 = 3.68 TO 2 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 2348
TO HOW MANY PLACES? 0

 2348 = 2348 TO 0 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 2348.6
TO HOW MANY PLACES? -1

 2348.6 = 2350 TO-1 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 2348.6
TO HOW MANY PLACES? -2

 2348.6 = 2300 TO-2 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 2348.6
TO HOW MANY PLACES? -3

 2348.6 = 2000 TO-3 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 2348.6
TO HOW MANY PLACES? -4

 2348.6 = 0 TO-4 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? -2.3469
TO HOW MANY PLACES? 3

-2.3469 =-2.347 TO 3 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF? 2
TO HOW MANY PLACES? 4

 2 = 2 TO 4 DECIMAL PLACES.

WHAT NUMBER DO YOU WISH TO ROUND OFF?
?C
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Biology NUMBER: NAME: EVOLU
SOURCE: Huntington Computer Project LANGUAGE: BASIC

DESCRIPTION: The EVOLU program demonstrates the effects of genetic mutations of a sample of 10000 pepper moths. These mutations cause a small number of these moths to become somewhat darker in tone. However, it takes many years for this effect to show up, making it impossible to show in a school lab. In addition, the program allows the student to vary the mutation rate and to change the environment midway in the experiment.

```
210 PRINT
215 PRINT "ENTER RATE OF MUTATION (1 TO 10)";
220 INPUT M
250 LET P0=10000
260 LET Z=P0
450 PRINT "YEAR IN WHICH ENVIRONMENTAL CHANGE OCCURS";
460 INPUT X
464 LET L=1
466 LET D=2
470 PRINT "CHANGE FAVORS LIGHT MOTHS OR DARK (ENTER 1 OR 2);
480 INPUT E
580 PRINT
582 PRINT "YEAR","DARK MOTHS","LIGHT MOTHS"
584 PRINT
610 FOR T=1 TO 30
620 IF T>=X THEN 650
630 LET P1=0
640 GO TO 710
650 IF E<>2 THEN 630
660 LET P1=INT(P1+.01*M*P0+.5)
670 LET P0=INT(Z-P1+.5)
680 IF P1<Z THEN 710
690 LET P1=Z
700 LET P0=0
710 PRINT T,P1,P0
720 NEXT T
999 END

READY
```

USAGE: Type RUN. When asked, enter the mutation rate (from 1 to 10), the year in which the environment is to change (3 to 10) and whether the environmental change will favor light (type a 1) or dark ones (type a 2). The program then prints out the course of evolution over a thirty year period.

```
ENTER RATE OF MUTATION (1 TO 10)? 2
YEAR IN WHICH ENVIRONMENTAL CHANGE OCCURS? 3
CHANGE FAVORS LIGHT MOTHS OR DARK (ENTER 1 OR 2)? 2

YEAR            DARK MOTHS      LIGHT MOTHS

 1              0               10000
 2              0               10000
 3              200             9800
 4              396             9604
 5              588             9412
 6              776             9224
 7              960             9040
 8              1141            8859
 9              1318            8682
 10             1492            8508
 11             1662            8338
 12             1829            8171
 13             1992            8008
 14             2152            7848
 15             2309            7691
 16             2463            7537
 17             2614            7386
 18             2762            7238
 19             2907            7093
 20             3049            6951
 21             3188            6812
 22             3324            6676
 23             3458            6542
 24             3589            6411
 25             3717            6283
 26             3843            6157
 27             3966            6034
 28             4087            5913
 29             4205            5795
 30             4321            5679
```

EDUCATIONAL BENEFITS: Where else can a student observe a thirty year reproductive cycle and watch the effects of mutations? More important, where else can the student repeat the cycle, trying new parameters and conditions?

Many other common science experiments take a long time to run, or require the student to run it several times. To get meaningful results, they may often by simulated on the computer.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Physics NUMBER: NAME: KINEMA
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: The KINEMA program tests the student's knowledge of kinematics. A ball, or any other object, is thrown up at a certain velocity (in meters per second.) The student is then asked to calculate certain information about the resulting trajectory. The computer then checks the response to see if it is within 15% of the correct answer.

```
100 PRINT
105 PRINT
106 LET Q=0
110 LET V=5+INT(35*RND(0))
111 PRINT "A BALL IS THROWN UPWARDS AT"V"METERS PER SECOND"
 112 PRINT
115 LET A=.05*V↑2
116 PRINT "HOW HIGH WILL IT GO (IN METERS)";
117 GOSUB 500
120 LET A=V/5
122 PRINT "HOW LONG UNTIL IT RETURNS (IN SECONDS)";
124 GOSUB 500
130 LET T=1+INT(2*V*RND(1))/10
132 LET A=V-10*T
134 PRINT "WHAT WILL ITS VELOCITY BE AFTER"T"SECONDS";
136 GOSUB 500
140 PRINT
150 PRINT Q"RIGHT OUT OF 3.";
160 IF Q<2 THEN 100
170 PRINT "  NOT BAD."
180 GO TO 100
500 INPUT G
502 IF ABS((G-A)/A)<.15 THEN 510
504 PRINT "NOT EVEN CLOSE...."
506 GO TO 512
510 PRINT "CLOSE ENOUGH."
511 LET Q=Q+1
512 PRINT "CORRECT ANSWER IS"A
520 PRINT
530 RETURN
999 END

READY
```

USAGE: Load KINEMA, then type RUN. Then try to answer the questions. Distances should be entered in meters, time in seconds, and velocity in meters per second (plus for up, minus for down.)

```
RUN


A BALL IS THROWN UPWARDS AT 13 METERS PER SECOND

HOW HIGH WILL IT GO (IN METERS)? 8
CLOSE ENOUGH.
CORRECT ANSWER IS 8.45

HOW LONG UNTIL IT RETURNS (IN SECONDS)? 3
NOT EVEN CLOSE....
CORRECT ANSWER IS 2.6

WHAT WILL ITS VELOCITY BE AFTER 1.7 SECONDS? -4
CLOSE ENOUGH.
CORRECT ANSWER IS-4


 2 RIGHT OUT OF 3.  NOT BAD.


A BALL IS THROWN UPWARDS AT 30 METERS PER SECOND

HOW HIGH WILL IT GO (IN METERS)? 50
CLOSE ENOUGH.
CORRECT ANSWER IS 45

HOW LONG UNTIL IT RETURNS (IN SECONDS)? 6
CLOSE ENOUGH.
CORRECT ANSWER IS 6

WHAT WILL ITS VELOCITY BE AFTER 2.8 SECONDS? 4
NOT EVEN CLOSE....
CORRECT ANSWER IS 2


 2 RIGHT OUT OF 3.  NOT BAD.
```

EDUCATIONAL BENEFIT: The key educational benefit here is the powerful motivating force that the computer has. The work that the student is being asked to do is really no different than rote homework exercises. But the computer makes it a challenge. It is very frustrating to have a dumb computer keep telling you you're wrong. Try it yourself and see if it doesn't make you want to brush up on your physics. That's real motivation.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Games/Physics NUMBER: NAME: LUNAR2
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC
(EduSystem 10 only)

DESCRIPTION: This program is a compressed version of the lunar module landing program originally written in 8K FOCAL. The program is also known as LUNAR, ROCKES, and ROCKET.

The program represents an exact simulation of an Apollo lunar landing module during the final descent. This portion of the descent would normally be controlled by the on-board computer backed up by a computer located on Earth. However, to exercise your knowledge of physics (and make an interesting game), both computers simultaneously have had a malfunction; therefore, you are on your own to safely land the spacecraft.

To make a soft landing, you may reset the burn rate of the retro rockets every ten seconds. You have a choice of not firing at all (burn rate=0) or firing at a rate between 8 and 200 lbs. per second. You have 16500 lbs. of fuel. If the rockets were not fired, your estimated free fall impact time is 120 seconds. The capsule weight is 33,000 pounds.

USAGE: This program runs on EduSystem 10 (4K) BASIC only. Retain only the SQR and RND functions (Option F). Type SCR to get rid of any existing programs and load LUNAR2. Type RUN to commence landing procedure. When a "?" is typed under burn rate, enter your burn rate and hit the return key. Remember, the only acceptable burn rates are 0 or any number between 8 and 200.

After you're down and the computer has typed "READY", if you wish to try it again, simply type RUN. Good luck!

```
LIST

13 PRINT "SEC","MI + FT","MPH","LB FUEL","BURN RATE"\PRINT
15 A=120\V=1\M=33000\N=16500\G=1.000000E-03\Z=1.8
21 PRINT L,INT(A);INT(5280*(A-INT(A))),3600*V,M-N,\INPUT K\T=10\PRINT
31 IF M-N<.001 THEN 41\IF T<.001 THEN 21\S=T\IF M<N+S*K THEN S=(M-N)/K
35 GOSUB 91 \IF I<=0 THEN 71 \IF V<=0 THEN 38 \IF J<0 THEN 81
38 GOSUB 61 \GO TO 31
41 PRINT "FUEL OUT AT";L;"SEC"\S=(-V+SQR(V*V+2*A*G))/G\V=V+G*S\L=L+S
51 W=3600*V\PRINT "ON MOON AT";L;"SEC - IMPACT VELOCITY";W;"MPH"
53 IF W>10 THEN 56 \PRINT "GOOD LANDING"\STOP
56 IF W>60 THEN 58 \PRINT "BAD JOLT"\STOP
58 PRINT "NO SURVIVORS-BLASTED NEW LUNAR CRATER"W*.2777;"FT DEEP"\STOP
61 L=L+S\T=T-S\M=M-S*K\A=I\V=J\RETURN
71 IF S<5.000000E-03 THEN 51 \D=V+SQR(V*V+2*A*(G-Z*K/M))\S=2*A/D
73 GOSUB 91 \GOSUB 61 \GO TO 71
81 W=(1-M*G/(Z*K))/2\S=M*V/(Z*K*(W+SQR(W*W+V/Z)))+.05\GOSUB 91
83 IF I<=0 THEN 71 \GOSUB 61 \IF J>0 THEN 31 \IF V>0 THEN 81 \GO TO 31
91 Q=S*K/M\J=V+G*S+Z*(-Q-Q*Q/2-Q↑3/3-Q↑4/4-Q↑5/5)
94 I=A-G*S*S/2-V*S+Z*S*(Q/2+Q↑2/6+Q↑3/12+Q↑4/20+Q↑5/30)\RETURN
99 END

RUN
SEC          MI + FT        MPH          LB FUEL        BURN RATE

 0            120  0         3600          16500          ?0
 10           109   5016     3636          16500          ?0
 20           99   4224       3672          16500          ?0
 30           89   2904       3708          16500          ?0
 40           79   1056       3744          16500          ?0
 50           68   3960       3780          16500          ?0
 60           58   1056       3816          16500          ?200
 70           48   564        3446.868      14500          ?200
 80           39   399        3050.708      12500          ?200
 90           31   987        2623.654      10500          ?0
 100          23   4483       2659.654      10500          ?200
 110          17   501        2196.947      8500           ?0
 120          10   4975       2232.947      8500           ?200
 130          5   2268        1728.634      6500           ?200
 140          1   2028        1175.137      4500           ?200

ON MOON AT 144.8262 SEC - IMPACT VELOCITY 887.6062 MPH
NO SURVIVORS-BLASTED NEW LUNAR CRATER 246.4883 FT DEEP

READY.
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Physics NUMBER: NAME: PHOTOE
SOURCE: Huntington Computer Project LANGUAGE: BASIC

DESCRIPTION: When light of short wave-length falls on a metal surface, electrons are ejected from the metal. According to the description of this phenomenon by Einstein, there is a maximum wavelength for each metal above which no electrons are emitted. This program lets the user find this critical wave-length.

APPROACH: The program simulates the actual experiment, where the metal is placed in a vacuum and bombarded with soft X-rays. The number of electrons ejected is collected and counted by means of an ammeter. The program types out three ammeter readings for each wave-length. After this has been done, the intensity may be increased and the experiment rerun. The typed out values coule then be graphed to show the critical wave-length.

To pick a metal enter its coefficient as follows:

Silver	.308
Bismuth	.338
Cadmium	.318
Lead	.340
Platinum	.385

EDUCATIONAL BENEFITS: Students very rarely get to work with the same concepts that Einstein worked with, or indeed with any of the great 20th century work in physics. The reason is simple: modern physics uses laboratory gear that is way beyond the means of most schools. The concepts themselves are often easy enough for the student; it is only the lack of equipment that holds him down. As this program demonstrates, the computer can fill this gap by simulating equipment. And unlike most lab equipment which is used with one unit, then lies idle the rest of the term, a computer does all kinds of things all year long. Its flexibility means it is never idle.

```
 100 PRINT
 110 PRINT "ENTER CO-EFFICIENT FOR METAL";
 120 INPUT V0
 460 LET K=INT(1+2*RND(X))
 470 PRINT
 480 PRINT " ","MEASURED CURRENT (MICRO-AMPERES)"
 490 PRINT "WAVELENGHTH","TRIAL 1","TRIAL 2","TRIAL 3"
 500 FOR L=.42 TO .25 STEP -.02
 510 LET M=INT(1000/L)
 520 PRINT M,
 530 FOR J=1 TO 3
 540 IF L>V0 THEN 570
 550 LET I=SQR(INT(25*RND(X)))
 560 GO TO 580
 570 LET I=SQR(K*K*100+INT(35*RND(X)))
 580 LET N=INT(10*I+.5)/10
 590 PRINT N,
 600 NEXT J
 610 PRINT
 620 NEXT L
 625 PRINT
 630 PRINT "INCREASE LIGHT INTENSITY (1 FOR YES, 2 FOR NO)";
 660 INPUT Q2
 670 IF Q2=2 THEN 999
 690 PRINT "BY WHAT FACTOR (1 TO 10)";
 700 INPUT F
 720 LET K=K*F
 730 GO TO 470
 999 END

READY

RUN


ENTER CO-EFFICIENT FOR METAL? .340

                MEASURED CURRENT (MICRO-AMPERES)
WAVELENGHTH     TRIAL 1         TRIAL 2         TRIAL 3
 2380            10.5            11.2            10.5
 2500            10.5            10              10.8
 2631            10.8            10.8            10
 2777            10.4            10.5            10.3
 2941            4.7             2.4             4.9
 3125            2.4             0               4.6
 3333            4.8             4.5             2.6
 3571            3.3             3.3             2.2
 3846            4.4             4               4.4

INCREASE LIGHT INTENSITY (1 FOR YES, 2 FOR NO)? 1
BY WHAT FACTOR (1 TO 10)? 5

                MEASURED CURRENT (MICRO-AMPERES)
WAVELENGHTH     TRIAL 1         TRIAL 2         TRIAL 3
 2380            50              50.1            50.2
 2500            50.2            50.2            50.2
 2631            50.2            50.2            50.3
 2777            50.2            50.2            50.3
 2941            4.8             3.7             1.4
 3125            3.2             3.7             3.7
 3333            2               1.4             4.6
 3571            1.7             3               4.2
 3846            2.8             1               4.1

INCREASE LIGHT INTENSITY (1 FOR YES, 2 FOR NO)? 2
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Physics/Electricity NUMBER: NAME: UELEC
SOURCE: J. Martin, Rutland High School LANGUAGE: BASIC

DESCRIPTION: This is a program which produces a table of values for the electric potential energy, in joules, at various separations, in meters (R), of two electric charges, Q1 and Q2.

$$U_{elec} = (K*Q1*Q2)/R$$

The user inputs the values of Q1, Q2, R1, R2 and the number of subdivisions desired in the total interval of distance R2-R1. Note that R1 must be less than R2.

```
  5 PRINT "PROGPAM UELEC"
  6 PRINT
100 PRINT "WHAT IS THE VALUE OF Q(1)";
110 INPUT Q1
120 PRINT "WHAT IS THE VALUE OF Q(2)";
130 INPUT Q2
140 PRINT
150 LET K=8.988 E9
160 PRINT "WHAT ARE THE DISTANCE LIMITS, R(1), R(2), BETWEEN"
 170 PRINT "WHICH YOU DESIRE THE VALUES TO RUN";
180 INPUT R1,R2
190 PRINT "HOW MANY SUBDIVISIONS OF THE INTERVAL DO YOU WISH";
200 INPUT S
205 PRINT
210 PRINT "DISTANCE, R","ELECTRIC POTENTIAL, U"
215 PRINT "   METERS","      JOULES"
216 PRINT
220 FOR R=R1 TO R2 STEP (R2-R1)/S
225 LET U=K*Q1*Q2/R
230 PRINT "   ";R,"    ";U
250 NEXT R
999 END

READY
```

```
RUN

PROGRAM UELEC

WHAT IS THE VALUE OF Q(1)? 10
WHAT IS THE VALUE OF Q(2)? 1

WHAT ARE THE DISTANCE LIMITS, R(1), R(2), BETWEEN
WHICH YOU DESIRE THE VALUES TO RUN? 1,2
HOW MANY SUBDIVISIONS OF THE INTERVAL DO YOU WISH? 12

DISTANCE, R    ELECTRIC POTENTIAL, U
   METERS              JOULES

     1                8.988000E+10
     1.083333         8.296615E+10
     1.166667         7.704000E+10
     1.25             7.190400E+10
     1.333333         6.741000E+10
     1.416667         6.344471E+10
     1.5              5.992000E+10
     1.583333         5.676632E+10
     1.666667         5.392800E+10
     1.75             5.136000E+10
     1.833333         4.902546E+10
     1.916667         4.689391E+10
     2                4.494000E+10

READY

RUN

PROGRAM UELEC

WHAT IS THE VALUE OF Q(1)? 10
WHAT IS THE VALUE OF Q(2)? 5

WHAT ARE THE DISTANCE LIMITS, R(1), R(2), BETWEEN
WHICH YOU DESIRE THE VALUES TO RUN? -
MORE? .5,5
HOW MANY SUBDIVISIONS OF THE INTERVAL DO YOU WISH? 9

DISTANCE, R    ELECTRIC POTENTIAL, U
   METERS              JOULES

     .5               8.988000E+11
     1                4.494000E+11
     1.5              2.996000E+11
     2                2.247000E+11
     2.5              1.797600E+11
     3                1.498000E+11
     3.5              1.284000E+11
     4                1.123500E+11
     4.5              9.986667E+10
     5                8.988000E+10


READY

-
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Electronics NUMBER: NAME: INDUC
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program combined with the indicated laboratory measurements will determine the value of an unknown inductor, such as a choke or transformer winding.

APPROACH: The impedance of an inductor is made up of two quantities; inductive reactance (X_L) and d.c. resistance (R). First, measure the resistance with an ohmmeter.

Next, determine the voltage (E) and current (I) of an a.c. circuit where the unknown inductance constitutes the entire load. Ohm's Law may be applied to find the impedance, $Z = E/I$.

We can now separate R from X_L by means of a vector diagram in which R becomes the base and Z, the hypotenuse. Once X_L has been obtained, the inductance L may be found from the formula:

$$L = \frac{X_L}{2\pi f}$$

In the formula, f is the frequency. A simple computer program can replace the various calculations and vector diagram. The program plus a sample run are shown on the back.





```
  5 PRINT "PROGRAM INDUC"
  6 PRINT
 10 PRINT "TO FIND THE VALUE OF AN UNKNOWN INDUCTOR,"
 11 PRINT "INPUT THE FOLLOWING QUANTITIES:"
 12 PRINT
 15 PRINT "RESISTANCE";
 16 INPUT R
 20 PRINT "VOLTAGE";
 21 INPUT E
 25 PRINT "CURRENT";
 26 INPUT I
 30 PRINT "FREQUENCY";
 31 INPUT Q
 35 LET Z=E/I
 40 LET X=SQR(Z*Z-R*R)
 45 LET L=X/(2*3.14159*Q)
 50 PRINT
 51 PRINT "INDUCTANCE IS"L"HENRIES."
 99 END

READY

RUN

PROGRAM INDUC

TO FIND THE VALUE OF AN UNKNOWN INDUCTOR,
INPUT THE FOLLOWING QUANTITIES:

RESISTANCE? 800
VOLTAGE? 100
CURRENT? .05
FREQUENCY? 60

INDUCTANCE IS 4.862268 HENRIES.


READY

RUN

PROGRAM INDUC

TO FIND THE VALUE OF AN UNKNOWN INDUCTOR,
INPUT THE FOLLOWING QUANTITIES:

RESISTANCE? 400
VOLTAGE? 115
CURRENT? .046
FREQUENCY? 60

INDUCTANCE IS 6.546029 HENRIES.


READY
```

$0.75

DIGITAL EQUIPMENT CORPORATION

education

Basic Applications Programs (Set 3)

BASIC APPLICATION PROGRAMS

- Set 3 -

The programs contained in this series are designed to demonstrate how the computer can be applied in a meaningful way to problems of many disciplines. The problems and the corresponding programs are, for the most part, quite simple and are designed to be "jumping off points" for students from the high school level on up.

All of the programs, with very few exceptions, may be run on EduSystems 10 through 90, PDP-8 family and PDP-11 family computers. Exceptions are noted in the program descriptions.

DEC welcomes contributions of programs and write-ups for use in this series. Contributors will be credited as program "Source." Please send programs and descriptions to Educational Marketing (5-2).

Additional copies of BASIC Applications Programs may be obtained from:

Direct Mail Department
Digital Equipment Corporation
Maynard, Massachusetts 01754

Price=	Set 1 (15 Programs)	$1.00
	Set 2 (19 Programs)	1.25
	Set 3 (12 Programs)	.75

BASIC APPLICATIONS PROGRAMS
Table of Contents (Set 3)

DIGITAL EQUIPMENT CORPORATION

education

AREA: General NUMBER: NAME: CAI-ADD
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program demonstrates a simplified Computer Assisted Instruction drill and practice routine. The user determines how many digits each problem will contain. Then, ten addition problems are produced with the user providing the answer after each problem is presented. After the last problem is correctly answered, the score is printed with an appropriate comment.

```
  5 REM - CAI ADDITION, 10 PROBLEMS
 10 PRINT "HOW MANY DIGITS";
 20 INPUT N
 22 LET W=0
 25 FOR X=1 TO 10
 27 LET Q=0
 30 LET A=INT(10↑N*RND(0))
 40 LET B=INT(10↑N*RND(0))
 50 PRINT
 60 PRINT "   ";A
 70 PRINT " +";B
 80 PRINT " -----"
 90 PRINT "  ";
100 INPUT G
110 IF G=A+B THEN 140
115 GOSUB 200
120 PRINT "WHAT?? TRY AGAIN"
130 GO TO 50
140 PRINT "THAT'S RIGHT. TRY ANOTHER"
150 NEXT X
155 PRINT
160 PRINT "YOU GOT"10-W"CORRECT THE FIRST TIME"
165 IF W<=3 THEN 180
170 PRINT "BUT YOU MISSED"W
175 STOP
180 PRINT "GOOD WORK!"
185 STOP
200 LET Q=Q+1
205 IF Q>1 THEN 120
210 LET W=W+1
215 RETURN
999 END
```

```
RUN

HOW MANY DIGITS? 2

   24
 + 29
 -----
 ? 52
WHAT?? TRY AGAIN

   24
 + 29
 -----
 ? 53
THAT'S RIGHT. TRY ANOTHER

   72
 + 31
 -----
 ? 103
THAT'S RIGHT. TRY ANOTHER

   30
 + 4
 -----
 ? 34
THAT'S RIGHT. TRY ANOTHER

   48
 + 49
 -----
 ? 87
WHAT?? TRY AGAIN

   48
 + 49
 -----
 ? 97
THAT'S RIGHT. TRY ANOTHER

   50
 + 4
 -----
 ? 54
THAT'S RIGHT. TRY ANOTHER

   23
 + 30
 -----
 ? 53
THAT'S RIGHT. TRY ANOTHER

   19
 + 91
 -----
 ? 100
WHAT?? TRY AGAIN

   19
 + 91
 -----
 ? 110
THAT'S RIGHT. TRY ANOTHER

   24
 + 98
 -----
 ? 112
WHAT?? TRY AGAIN

   24
 + 98
 -----
 ? 122
THAT'S RIGHT. TRY ANOTHER

   25
 + 3
 -----
 ? 28
THAT'S RIGHT. TRY ANOTHER

   87
 + 92
 -----
 ? 179
THAT'S RIGHT. TRY ANOTHER

YOU GOT 6 CORRECT THE FIRST TIME
BUT YOU MISSED 4
```

EDUCATIONAL BENEFIT: This program could be used for drill and practice in a math class. However, it was designed to merely demonstrate the basic idea behind a traditional Computer Assisted Instruction program. It may be used to give students some clues for writing more sophisticated CAI programs.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: DISTANCE
SOURCE: Digital Equipment Corp. LANGUAGE: BASIC

DESCRIPTION: This program solves for the distance between two points in 3-dimensional space defined by their x, y, z coordinates.

APPROACH: The distance between two points in 3-dimensional space may be found by the formula:

$$d = (x_2-x_1)^2 + (y_2-y_1)^2 + (z_2-z_1)^2$$

Program distance solves the equation for three sets of points with the following coordinates:

P (0,0,0)	and Q (3, 4, 5)
A (3.5, -4.7, 6.2)	and B (0.9, 3.0, 4.4)
X (67, 36, 82	and Y (54, 25, 90)

Notice how we have used a DATA statement to define the points. Is this a good "general" approach? Modify the program to accept any two sets of x, y, z coordinates at the time the program is run and output the distance.

This calculation is used extensively in aerospace navigation. A good exercise for advanced students is determining the angle or "compass heading" of the resulting path. Better start first with just two dimensions!

```
10 FOR N=1 TO 3
15 PRINT
20 READ A,B,C,D,E,F
30 LET D1=SQR((A-D)↑2+(B-E)↑2+(C-F)↑2)
40 PRINT "THE DISTANCE BETWEEN ("A","B","C") AND ("D","E","F")"
45 PRINT "IS"D1
50 NEXT N
60 DATA 0,0,0,3,4,5,3.5,-4.7,6.2,-.9,3,4.4,67,36,82,54,25,90
99 END

RUN


THE DISTANCE BETWEEN ( 0 , 0 , 0 ) AND ( 3 , 4 , 5 )
IS 7.071068

THE DISTANCE BETWEEN ( 3.5 ,-4.7 , 6.2 ) AND (-.9 , 3 , 4.4 )
I S 9.049309

THE DISTANCE BETWEEN ( 67 , 36 , 82 ) AND ( 54 , 25 , 90 )
I S 18.81489


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Mathematics NUMBER: NAME: EXPON
SOURCE: NREL/SDC LANGUAGE: BASIC

DESCRIPTION: Program EXPON solves for x in any exponential equation of the form:

$$A^{mx+n} = B$$

given the values of A, B, m, and n.

The program solves the following problems:

1. $5^{x} = 50$
2. $5^{3x+1} = 7.6$
3. $17^{x-3} = 8.12$
4. $11^{1-2x} = 247$

APPROACH: If you were to solve an exponential equation by hand, you would probably go through the following steps:

$$5^{x} = 40$$

$$\log 5^{x} = \log 40$$

$$x \log 5 = \log 40$$

$$x = \frac{\log 40}{\log 5}$$

$$x = \frac{1.6021}{.6990} = 2.292$$

However, in the generalized form, the solution for x is:

$$x = \frac{\frac{\log B}{\log A}}{M-(N/M)}$$

A DATA statement was used to contain the data to solve problems 1-4 above. Modify the program to accept data using an INPUT statement to solve any exponential equation.

```
10 READ A,B,M,N
20 LET X=(LOG(B)/LOG(A))/M-(N/M)
30 PRINT X
40 GOTO 10
50 DATA 5,40,1,0,5,7.6,3,1,17,8.12,1,-3,11,247,-2,1
99 END

RUN

 2.29203
 .0867198
 3.739207
-.6487967

OUT OF DATA IN LINE 10


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Chemistry NUMBER: NAME: GASVOL
SOURCE: NREL/SDC/DEC LANGUAGE: BASIC

DESCRIPTION: The volume of a gas varies directly with the absolute temperature T (Kelvin) and inversely with the pressure P. If a certain quantity of gas occupies 500 cubic feet at a pressure of 53 pounds per square foot and an absolute temperature of 500 degrees, what volume will it occupy at 600 degrees absolute temperature and pressures from 100 to 1000 pounds per square foot in increments of 50 pounds?

The original conditions are used to solve for the constant K in line 10. (K = VP/T). Then the new volumes are computed for varying pressures with T = 600° using the formula V = KT/P.

In the second part of the program lines 70 - 130 have been added to give a plot of the volume for the various pressures.

How would you modify the program to deal with a more general case (i.e., other gasses and different temperatures)?

```
10 LET K=500*53/500
20 PRINT "A GAS, AT 600 DEGREES KELVIN, AND PRESSURE AS"
25 PRINT "LISTED, OCCUPIES THE VOLUME SPECIFIED."
30 PRINT
40 PRINT "PRESSURE","VOLUME"
50 FOR P=100 TO 1000 STEP 50
60 LET V=K*600/P
70 PRINT P,V
80 NEXT P
99 END
```

```
RUN

A GAS, AT 600 DEGREES KELVIN, AND PRESSURE AS
LISTED, OCCUPIES THE VOLUME SPECIFIED.

PRESSURE        VOLUME
 100             318
 150             212
 200             159
 250             127.2
 300             106
 350             90.85714
 400             79.5
 450             70.66667
 500             63.6
 550             57.81818
 600             53
 650             48.92308
 700             45.42857
 750             42.4
 800             39.75
 850             37.41176
 900             35.33333
 950             33.47368
 1000            31.8


READY

  70 PRINT P;
  80 LET X=INT(V*2/10+.5)
  90 FOR L=1 TO X
 100 PRINT " ";
 110 NEXT L
 120 PRINT "X"
 130 NEXT P
 999 END

READY

RUN

A GAS, AT 600 DEGREES KELVIN, AND PRESSURE AS
LISTED, OCCUPIES THE VOLUME SPECIFIED.

PRESSURE        PLOT OF VOLUME
 100                                                          X
  150                                                    X
 200                                           X
 250                                      X
 300                                  X
 350                              X
 400                          X
 450                       X
 500                      X
 550                    X
 600                   X
 650                 X
 700               X
 750              X
 800              X
 850            X
 900            X
 950            X
 1000           X
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Physics NUMBER: NAME: NEWTON
SOURCE: Project SOLO, Pittsburgh, Pa. LANGUAGE: BASIC

DESCRIPTION: (This problem assumes the student has basic knowledge of Newton's three laws)

The first problem we will solve is as follows:
A car at rest at a traffic light weighs 5.63×10^3 newtons.
(a) What is the mass of the car?
When the light changes to green, it accelerates at 3.18 m/sec^2.
(b) If friction is negligible, what must be the force between the rear wheels and the road?
(c) What is the velocity of the car when it has moved 30 meters?

Remembering that gravitational force is 9.8 m/sec^2 and W=MA, we can calculate the mass of the car directly in statement 20. We can then make use of the relationships:

$$F = MA$$
$$V = V_o + A * T \text{ or } = V_o + 2 * A * D$$

```
 5 PRINT "INPUT W, A, D":
10 INPUT W,A,D
20 LET M=W/9.8
30 LET F=M*A
40 LET V=2*A*D
50 PRINT "M="M,"F="F,"V="V
99 END

READY

RUN

INPUT W, A, D? 5.63E3, 3.18, 30
M= 574.4898    F= 1826.878    V= 190.8
```

If some other force in addition to the gravitational force acts on a falling body, it's acceleration will not be 9.8 m/sec^2. Consider the following problem: The Saturn V rocket has a mass of 3.18×10^6kg. The thrust of its first stage engines is quoted as being 7.5×10^6lbs. What is the initial acceleration of the rocket as it blasts off for the moon.

In this case, Newton's second law is expressed:

$$(F + W) = MA$$

Since the direction of rocket acceleration is upward, W has a negative value.

```
5 PRINT "INPUT M, F";
10 INPUT M,F
20 LET W=-M*9.8
30 LET A=(F+W)/M
40 PRINT "ACCEL. ="A
99 END

RUN

INPUT M, F? 3.18E6, 3.34E7
ACCEL = .7031447
```

FOR MORE INFORMATION: The teaching module "Newton's Laws of Motion" containing six additional excellent problems is available complete from:

Dr. Thomas Dwyer
Project SOLO, Computation Center
University of Pittsburgh
Pittsburgh, Pa. 15213

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business NUMBER: NAME: FIFO
SOURCE: Gregg Accounting, Advanced Course LANGUAGE: BASIC
Gregg, McGraw-Hill, 2d edition (EduSystem 10)

DESCRIPTION: This program demonstrates the First-in, First-out (FIFO) method of inventory evaluation that is used by many businesses.

APPROACH: We read the beginning inventory from a DATA statement and put the quantity and price information into a simple array Q(X), P(X). Each time units are removed from inventory we cycle through the array, removing items from inventory in the same order they were placed into the inventory. When zero units remain at one price, the next set of units are taken at the next price. This program accepts six sets of data in Line 40.

EDUCATIONAL APPLICATIONS: Used to show how FIFO works. Change the price information in Line 40 (quantity, then price) to show the affect FIFO has on shipment values in terms of falling costs (example uses rising costs). Let your students rewrite the program for Last-in, First-out (LIFO) inventory evaluation.

*Program is coded using multiple statements per line feature of certain EduSystems.

```
LIST
1 REM ARK-FIFO
2 FOR X=1 TO 6\READ Q(X),P(X)\NEXT X\S=X-1
6 GOSUB 20 \PRINT "ENTER QUANTITY REMOVED";\INPUT W\PRINT "SHIPPED OUT"
8 PRINT "NUMBER","PRICE","TOTAL"
10 FOR X=1 TO S\IF Q(X)=0 THEN 18
12 IF Q(X)-W>=0 THEN 34
14 PRINT Q(X),P(X),(Q(X)*P(X))
16 T1=T1+(Q(X)*P(X))\W=W-Q(X)\Q(X)=0
18 NEXT X\IF Q(X+1)<=0 THEN 38
20 PRINT "REMAINING INVENTORY"\T=0
22 PRINT "NUMBER","PRICE","TOTAL"
24 FOR X=1 TO S\IF Q(X)<=0 THEN 30
26 PRINT Q(X),P(X),(Q(X)*P(X))
28 T=T+(Q(X)*P(X))
30 NEXT X
32 PRINT "TOTAL VALUE ="T\PRINT \RETURN
34 PRINT W,P(X),W*P(X)\T1=T1+(W*P(X))
36 PRINT "SHIPPED VALUE="T1\T1=0\Q(X)=Q(X)-W\GO TO 6
38 PRINT "OUT OF INVENTORY"\PRINT "SHIPPED VALUE ="T1
40 DATA 35,11.25,60,12,100,12.5,50,13,80,13.5,40,14.25\END
```

```
RUN
REMAINING INVENTORY
NUMBER          PRICE           TOTAL
 35              11.25           393.75
 60              12              720
 100             12.5            1250
 50              13              650
 80              13.5            1080
 40              14.25           570
TOTAL VALUE = 4663.75

ENTER QUANTITY REMOVED?100
SHIPPED OUT
NUMBER          PRICE           TOTAL
 35              11.25           393.75
 60              12              720
 5               12.5            62.5
SHIPPED VALUE= 1176.25
REMAINING INVENTORY
NUMBER          PRICE           TOTAL
 95              12.5            1187.5
 50              13              650
 80              13.5            1080
 40              14.25           570
TOTAL VALUE = 3487.5
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business/Math NUMBER: NAME: INDIAN
SOURCE: NREL/SDC LANGUAGE: BASIC

DESCRIPTION: This program solves a simple interest problem which makes one think twice about the deal that the Dutch made with the Indians. If the $24 which the Indians received in 1626 for Manhattan Island had been deposited in a bank paying 5-3/4% interest compounded annually, what would it have amounted to by 1971?

APPROACH: If P dollars are invested at an interest rate of r (expressed as a decimal) compounded n times, the total amount A is given by the formula:

$$A = P(1 + r)^n$$

How much was gained in 1971 alone?

```
10 LET P=24
20 LET R=.0575
30 LET N=1971-1626
40 LET A=P*((1+R)↑N)
50 PRINT "IF THE $24 WHICH THE INDIANS RECEIVED IN 1626 FOR"
 60 PRINT "MANHATTAN ISLAND HAD BEEN DEPOSITED IN A BANK"
 70 PRINT "PAYING"R*100"% INTEREST COMPOUNDED ANNUALLY, IT WOULD"
 80 PRINT "HAVE AMOUNTED TO $"A"BY 1971."
99 END

READY

RUN

IF THE $24 WHICH THE INDIANS RECEIVED IN 1626 FOR
MANHATTAN ISLAND HAD BEEN DEPOSITED IN A BANK
PAYING 5.75 % INTEREST COMPOUNDED ANNUALLY, IT WOULD
HAVE AMOUNTED TO $ 5.713946E+9 BY 1971.


READY
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business, Math NUMBER: NAME: INT-1
SOURCE: CAMP Algebra, DEC LANGUAGE: BASIC

DESCRIPTION: This program illustrates how interest accumulates using compound interest methods as compared to using simple interest.

Input principal, interest rate, and number of times compounded each year. For each year, the program calculates interest earnings.

(1) assuming simple interest.
(2) assuming interest is compounded.

The computer will print your year end account balance each year for 10 years. You should use even dollar figures for the principal amount to make it easier to "see" the difference.

```
 5 REM   PROGRAM INT-1
10 REM   SIMPLE AND QUARTERLY COMPUUND INTEREST
14 PRINT "WHAT IS THE PRINCIPAL AMOUNT";
18 INPUT P
22 PRINT "WHAT IS THE RATE OF INTEREST (X.XX)";
26 INPUT R
27 PRINT "COMPOUNDED HOW MANY TIMES PER YEAR";
28 INPUT N
30 LET R=R/100
34 LET A=P*R
38 LET S=P+A
40 PRINT
42 PRINT "YEAR","SIMPLE","COMPOUND"
46 FOR Y=1 TO 10
50 FOR X=1 TO N
54 LET P=P*(1+R/N)
58 NEXT X
62 PRINT Y,S,P
66 LET S=S+A
70 NEXT Y
74 PRINT
78 GOTO 14
99 END
```

```
RUN

WHAT IS THE PRINCIPAL AMOUNT? 1000
WHAT IS THE RATE OF INTEREST (X.XX)? 5.25
COMPOUNDED HOW MANY TIMES PER YEAR? 4

YEAR            SIMPLE          COMPOUND
 1               1052.5          1053.543
 2               1105            1109.952
 3               1157.5          1169.382
 4               1210            1231.994
 5               1262.5          1297.958
 6               1315            1367.454
 7               1367.5          1440.671
 8               1420            1517.808
 9               1472.5          1599.075
 10              1525            1684.694

WHAT IS THE PRINCIPAL AMOUNT? 1000
WHAT IS THE RATE OF INTEREST (X.XX)? 5.25
COMPOUNDED HOW MANY TIMES PER YEAR? 365

YEAR            SIMPLE          COMPOUND
 1               1052.5          1053.895
 2               1105            1110.695
 3               1157.5          1170.555
 4               1210            1233.642
 5               1262.5          1300.129
 6               1315            1370.199
 7               1367.5          1444.046
 8               1420            1521.873
 9               1472.5          1603.894
 10              1525            1690.336

WHAT IS THE PRINCIPAL AMOUNT? 1
WHAT IS THE RATE OF INTEREST (X.XX)? 100
COMPOUNDED HOW MANY TIMES PER YEAR? 1000

YEAR            SIMPLE          COMPOUND
 1               2               2.716889

↑C

READY
```

EDUCATIONAL BENEFITS: The computer is used to demonstrate the advantages of interest compounding. By using different interest rates you can illustrate the importance of shopping around for a higher interest rate for your savings.

In the last example, we used a 100% (!) rate of interest and compounded it 1000 times per year. Try it compounding it 5000 times per year (run time will be in minutes). Does the compound interest amount begin to look like the constant "e"? Why?

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business NUMBER: NAME: INT-2
SOURCE: Huntington Computer Project LANGUAGE: BASIC

DESCRIPTION: This program will calculate the balance of a savings account in which deposits are made regularly. Pick an amount of money to save each month and pick how many years you will save it and this program will tell you how much you will have at the end of that time.

USAGE: In any class where thrift or savings is taught this program can be used to illustrate how quickly money can accumulate when you make regular savings deposits. You may also use this program to show how small differences in interest rates can make significant differences in your interest earnings.

```
4 PRINT "THIS PROGRAM WILL CALCULATE THE BALANCE OF A SAVINGS"
5 PRINT "ACCOUNT IN WHICH DEPOSITS ARE MADE REGULARLY."
6 PRINT
7 PRINT "HOW MUCH WILL YOU SAVE EACH MONTH";
8 INPUT A
10 PRINT "HOW MANY INTEREST COMP OUND PERIODS EACH YEAR";
12 INPUT B
14 PRINT "AT WHAT RATE OF INTEREST (XX.X%)";
16 INPUT C
18 PRINT "HOW MANY YEARS WILL YOU MAKE THIS SAVING";
19 INPUT D
20 LET F=0
21 LET A=A*(12/B)
22 LET E=(C/100)/B
24 LET G=B*D
26 LET T1=0
30  LET T1=T1+1
40 IF T1=G+1 THEN 130
50 LET F=(1+E)*(F+A)
60 GO TO 30
130 LET F=INT(100*F+.5)/100
135 PRINT
136 PRINT
140 PRINT "AFTER"D"YEARS YOU WILL HAVE $"F
145 PRINT
146 PRINT
150 GO TO 6
999 END
```

```
RUN

THIS PROGRAM WILL CALCULATE THE BALANCE OF A SAVINGS
ACCOUNT IN WHICH DEPOSITS ARE MADE REGULARLY.

HOW MUCH WILL YOU SAVE EACH MONTH? 100
HOW MANY INTEREST COMPOUND PERIODS EACH YEAR? 4
AT WHAT RATE OF INTEREST (XX.X%)? 5
HOW MANY YEARS WILL YOU MAKE THIS SAVING? 5


AFTER 5 YEARS YOU WILL HAVE $ 6853.5


HOW MUCH WILL YOU SAVE EACH MONTH? 20
HOW MANY INTEREST COMPOUND PERIODS EACH YEAR? 2
AT WHAT RATE OF INTEREST (XX.X%)? 5.5
HOW MANY YEARS WILL YOU MAKE THIS SAVING? 10


AFTER 10 YEARS YOU WILL HAVE $ 3230.14


HOW MUCH WILL YOU SAVE EACH MONTH? 10
HOW MANY INTEREST COMPOUND PERIODS EACH YEAR? 365
AT WHAT RATE OF INTEREST (XX.X%)? 5
HOW MANY YEARS WILL YOU MAKE THIS SAVING? 6


AFTER 6 YEARS YOU WILL HAVE $ 839.69


HOW MUCH WILL YOU SAVE EACH MONTH? 10
HOW MANY INTEREST COMPOUND PERIODS EACH YEAR? 1
AT WHAT RATE OF INTEREST (XX.X%)? 3.75
HOW MANY YEARS WILL YOU MAKE THIS SAVING? 6


AFTER 6 YEARS YOU WILL HAVE $ 820.63
```

DIGITAL EQUIPMENT CORPORATION

education

AREA: Business NUMBER: NAME: SORT-2
SOURCE: Computer Methods in Mathematics LANGUAGE: BASIC

DESCRIPTION: Given a list of eight numbers: 75, 83, 67, 52, 99, 66, 83, 72. Sort the numbers into increasing order, left to right.

52, 66, 67, 72, 75, 83, 83, 99

Let the computer do it. A program and RUN are shown on the following page. Examine them.

How does the program work? Add the following statements and RUN the program. The printout will show you the status of the list <u>after</u> Lines 420 through 480 have been carried out for I = 1, then I = 2 and so on.

```
473 PRINT "I=";I;"LIST IS";
477 GOSUB 800
505 PRINT
```

The original list of numbers is in Line 901. To sort a different list of eight numbers, simply change Line 901. Try it for the following lists, using the modified program, including Lines 473, 477, and 505.

(1) 2, 1, 3, 4, 5, 6, 7, 8 (2) 1, 2, 3, 4, 5, 6, 8, 7

(3) 4, 3, 2, 1, 5, 6, 7, 8, (4) 8, 7, 6, 5, 4, 3, 2, 1

If you need more evidence, delete Lines 473, 477, and 505, then add the following Lines and RUN the program for the above lists.

```
415 PRINT "FOR I=";I;"AND"
470 PRINT "J=";J;"LIST IS";
471 GOSUB 800
472 NEXT J
```

```
100 REMARK   SORT EIGHT NUMBERS
110 PRINT "THIS PROGRAM SORTS A LIST OF EIGHT NUMBERS INTO"
120 PRINT "INCREASING ORDER AND PRINTS THE FINAL LIST WITH"
130 PRINT "THE SMALLEST NUMBER FIRST AND LARGEST LAST"
140 PRINT
200 REMARK   READ EIGHT NUMBERS INTO X(1), X(2),...,X(8)
210 DIM X(10)
220 FOR K=1 TO 8
230 READ X(K)
240 NEXT K
300 REMARK   PRINT THE ORIGINAL (UNSORTED) LIST
310 PRINT "UNSORTED LIST    ";
320 GOSUB 800
330 PRINT
400  REMARK   SORT THE NUMBERS
410 FOR I=1 TO 7
420 FOR J=I+1 TO 8
430 IF X(I) <= X(J) THEN 470
440 LET T=X(I)
450 LET X(I)=X(J)
460 LET X(J)=T
470 NEXT J
480 NEXT I
500 REMARK   PRINT THE FINAL (SORTED) LIST
510 PRINT "SORTED LIST      ";
520 GOSUB 800
530 STOP
800 REMARK   SUBROUTINE TO PRINT X(1), X(2),...,X(8)
810 FOR K=1 TO 8
820 PRINT X(K);
830 NEXT K
840 PRINT
850 RETURN
900 REMARK   HERE IS THE ORIGINAL LIST OF NUMBERS
910 DATA 75,83,67,52,99,66,83,72
999 END

RUN

THIS PROGRAM SORTS A LIST OF EIGHT NUMBERS INTO
INCREASING ORDER AND PRINTS THE FINAL LIST WITH
THE SMALLEST NUMBER FIRST AND LARGEST LAST

UNSORTED LIST     75   83   67   52   99   66   83   72

SORTED LIST       52   66   67   72   75   83   83   99

READY
```

Note: If you have trouble fitting the program into EDUSYSTEM 10, delete lines 100, 110, 120, 130, 200, 300, 400, 500 and 900. But <u>don't</u> delete Line 800.

DIGITAL EQUIPMENT CORPORATION

education

AREA: Social Studies NUMBER: NAME: POPULATION
SOURCE: Gruenberger and Jaffray LANGUAGE: BASIC
Problems for Computer Solution, Wiley

DESCRIPTION: This program solves the following problem:
Assume that in 1960 the population figures for the United States and Mexico were 180,000,000 and 85,000,000 respectively. Assume also that the annual rate of growth for the United States in 1.23%, and that for Mexico, 2.23%. If these growth rates remain constant, in what year will the population of Mexico equal or exceed that of the United States

```
10 LET U=180000000
20 LET M=85000000
30 LET R1=1.0123
40 LET R2=1.0223
50 LET Y=1960
60 LET U=U*R1
70 LET M=M*R2
80 LET Y=Y+1
90 IF M<U THEN 60
100 PRINT "U.S. POPULATION IS"U"AND MEXICO IS"M"IN"Y
999 END

RUN

U.S. POPULATION IS 4.614066E+8 AND MEXICO IS 4.644644E+8 IN 2037
```

Modify the problem to take into account that the U.S. growth rate increases 0.01% each year and that of Mexico increases by 0.001% each year. Will the population of Mexico under these circumstances ever exceed that of the United States?

```
60 LET U=U*(1.01*R1)
70 LET M=M*(1.001*R2)
85 IF Y>10000 THEN 120
110 STOP
120 PRINT "U.S. STILL HAS MORE PEOPLE,"U"THAN MEXICO,"M"IN 10000."

RUN

???
```

Consider similar data as the first case for the U.S. (180,000,000) and California (15,700,000) for 1960, and growth rates of 1.23% and 3.70%. Applying the same routine, we would find a year in which the population of California exceeded the population of the United States, which is nonsense. Where is the discrepency?

Or try this version of the problem. In 1963, the United States consumption of motor fuel was 64,000,000 gallons, increasing by 4 percent per year. The consumption in California was 6,000,000, increasing 5.4 percent per year. In what year will California consume more than the United States.

EDUCATIONAL BENEFIT: The principle illustrated by the last two exercises is that of GIGO - "garbage in, garbage out." If the problem definition is intrinsically nonsensical, then the application of the computer to it will add no sense. Blaming the nonsensical result, then, on the computer is not fair, but we see examples of it frequently.

The last problem could make sense if we ask in what year the consumption of fuel in California exceeds that of the <u>remainder</u> of the United States, if ever?

DIGITAL EQUIPMENT CORPORATION

education

AREA: Games NUMBER: NAME: SLOTS
SOURCE: Anonomous LANGUAGE: BASIC

DESCRIPTION: This program is a simulation of a slot machine, or one-arm bandit, of the type popular at Las Vegas, Monte Carlo and other gambling spas.

It is not a perfect simulation since three bars or three $ are frequently, at the real casinos, a big ($5 or $10) jackpot. However, the operation of the slot machine as simulated by this program is probably more honest than most existing real machines. Calculate the average number of "pulls" needed to lose $20.00. What is the probability of winning $20.00?

The program is on the reverse. The instructions and two "pulls" are shown below.

```
READY

RUN
DO YOU WISH INSTRUCTIONS 1-YES, Ø-NO? 1
THIS IS A SIMULATION OF A SLOT MACHINE USING A COMPUTER
EACH TIME YOU 'PULL' I WILL ASK YOU IF YOU WISH TO PLAY AGAIN
IF YOU DO SIMPLY INPUT A'1' FOLLOWED BY A RETURN,
OTHERWISE INPUT A  'Ø' FOLLOWED BY A RETURN.
NOW WE ARE READY TO PLAY


          LEMON     LEMON     $
YOU HAVE LOST 1 DOLLAR -- TOTAL = $-1
                    DO YOU WISH TO PULL AGAIN? 1


          $         BELL      APPLE
YOU HAVE LOST 1 DOLLAR -- TOTAL = $-2
                    DO YOU WISH TO PULL AGAIN? Ø
IT'S BEEN NICE OPERATING FOR YOU COME BACK SOON!
```

```
READY

TAPE

   1 RANDOMIZE
   2 DIM D(3)
   5 PRINT"DO YOU WISH INSTRUCTIONS 1-YES, Ø-NO";
   6 INPUT J
   7 IF J=Ø THEN 13
   8 PRINT"THIS IS A SIMULATION OF A SLOT MACHINE USING A COMPUTER "
   9 PRINT"EACH TIME YOU 'PULL' I WILL ASK YOU IF YOU WISH TO PLAY AGAIN"

  10 PRINT"IF YOU DO SIMPLY INPUT A'1' FOLLOWED BY A RETURN,"
  11 PRINT"OTHERWISE INPUT A  'Ø' FOLLOWED BY A RETURN."
  12 PRINT"NOW WE ARE READY TO PLAY "
  13 PRINT
  14 PRINT
  15 PRINT TAB (9)" ";
 100 FOR B1=1 TO 3
 110 LET D(B1)=INT(RND(Ø)*6)+1
 120 NEXT B1
 130 FOR G1=1 TO 3
 140 IF D(G1)=1 THEN 390
 150 IF D(G1)=2 THE N 410
 160 IF D(G1)=3 THEN 430
 170 IF D(G1)=4 THEN 450
 180 IF D(G1)=5 THEN 470
 190 IF D(G1)=6 THEN 490
 385 GOTO 999
 390 PRINT"BELL     ";
 400 GOTO 500
 410 PRINT"BAR        ";
 420 GOTO 500
 430 PRINT "CHERRY    ";
 440 GOTO 500
 450 PRINT "APPLE     ";
 460 GOTO 500
 470 PRINT "LEMON     ";
 480 GOTO 500
 490 PRINT "$         ";
 500 NEXT G1
 510 IF D(1)<>D(2) THEN 520
 511 IF D(2)=D(3) THEN 600
 520 PRINT
 530 IF D(1)=D(3) THEN 550
 531 LET B=B-1
 532 PRINT "YOU HAVE LOST 1 DOLLAR -- TOTAL = $"B
 533 GOTO 800
 550 LET B=B+.50
 560 PRINT "YOU HAVE WON 50 CENTS -- TOTAL = $"B
 570 GOTO 800
 600 LET B=B+1
 610 PRINT"YOU HAVE WON 1 DOLLAR -- TOTAL $"B
 620 GOTO 800
 800 PRINT TAB(20)"DO YOU WISH TO PULL AGAIN";
 801 INPUT J
 802 IF J=1 THEN 13
 803 PRINT"IT'S BEEN NICE OPERATING FOR YOU COME BACK SOON!"
 999 END

READY
```